RollBasicAndGeneralized

Empirical Market Microstructure

(2006, Oxford University Press)

Companion *Mathematica* notebook

Joel Hasbrouck

This notebook covers material in Chapters 3, 4 and 8.

```
Text @ Style["Notebook evaluated " <> DateString["DateTime"], "Subtitle"]
```

Notebook evaluated Wednesday 13 June 2007 10:28:38

Preliminaries

```
<< Notation`
```

The following commands define symbolizations that are convenient for labeling rules.

```
Symbolize[Anything_Rule]; Symbolize[Anything_Rules];
```

Set up symbols that will be used as subscripted paramaters:

```
Symbolize[γ_];
Symbolize[σ²_];
```

Define an "expectations operator". This isn't a real expectations operator, but it knows some simple rules like linearity.
The following command allows the "expectation" operator to be entered as ESC E ESC.

```
AddInputAlias[ℰ, "E"]
```

```
ℰLinearity_Rules = {
   ℰ[a_ + b_] :> ℰ[a] + ℰ[b],
   ℰ[c_] :> c /; NumberQ[c],
   ℰ[c_Symbol] :> c /; MemberQ[Attributes[c], Constant],
   ℰ[c_ a_] :> c ℰ[a] /; NumberQ[c],
   ℰ[c_Symbol a_] :> c ℰ[a] /; MemberQ[Attributes[c], Constant],
   ℰ[c_Symbol^n_Integer a_] :> c^n ℰ[a] /; MemberQ[Attributes[c], Constant]};
```

```
SetAttributes[ℰ, Listable]
```

Basic Roll (Chapter 3)

Model:

```
m_Rule = m_t_ :> m_t-1 + u_t;
p_Rule = p_t_ :> m_t + c q_t;
Δp_Rule = Δp_t_ :> (p_t /. p_Rule /. m_Rule) - (p_t-1 /. p_Rule);
Δp_t /. Δp_Rule
```

$-c\, q_{-1+t} + c\, q_t + u_t$

To obtain $\text{Var}[\Delta p_t] = \gamma_0$, consider:

```
Expand[Δp_t^2 /. Δp_Rule]
```

$c^2\, q_{-1+t}^2 - 2\, c^2\, q_{-1+t}\, q_t + c^2\, q_t^2 - 2\, c\, q_{-1+t}\, u_t + 2\, c\, q_t\, u_t + u_t^2$

Now define some additional rules for the expectations operator. (These are specific to the model at hand.)

```
ℰRoll_Rules = {
   ℰ[q_^2] → 1,
   ℰ[u_^2] → σ_u^2,
   ℰ[q_ u_] → 0,
   ℰ[q_t_ q_s_] :> 0 /; t =!= s,
   ℰ[u_t_ u_s_] :> 0 /; t =!= s};
```

```
Attributes[c] = {Constant};
```

Then $\gamma_0 =$

```
ℰ[Expand[Δp_t^2 /. Δp_Rule]] //. ℰLinearity_Rules //. ℰRoll_Rules
```

$2\, c^2 + \sigma_u^2$

To obtain $\mathrm{Cov}[\Delta p_t, \Delta p_{t-1}] = \gamma_1$:

```
Expand[Δp_t Δp_{t-1} /. Δp_Rule]
```

$c^2 q_{-2+t} q_{-1+t} - c^2 q_{-1+t}^2 - c^2 q_{-2+t} q_t + c^2 q_{-1+t} q_t - c\, q_{-1+t} u_{-1+t} + c\, q_t u_{-1+t} - c\, q_{-2+t} u_t + c\, q_{-1+t} u_t + u_{-1+t} u_t$

```
ℰ[Expand[Δp_t Δp_{t-1} /. Δp_Rule]] //. ℰLinearity_Rules //. ℰRoll_Rules
```

$-c^2$

The second-order autocorrelation is ...

```
ℰ[Expand[Δp_t Δp_{t-2} /. Δp_Rule]] //. ℰLinearity_Rules //. ℰRoll_Rules
```

0

... and so on.

Elements of univariate time-series analysis (Chapter 4)

Moving average forms

```
Clear[θ]; γ0 =.; γ1 =.;
```

The MA(1) process is $x_t = \epsilon_t + \theta\epsilon_{t-1}$. The autocovariances are:

```
γ_Rules = {γ0 → (θ^2 + 1) σ_ϵ^2, γ1 → θ σ_ϵ^2};
γ_Rules // TableForm
```

$\gamma_0 \to (1+\theta^2)\,\sigma_\epsilon^2$

$\gamma_1 \to \theta\,\sigma_\epsilon^2$

To solve backwards from the autocovariances to the model parameters:

```
γEquations = Apply[Equal, γ_Rules, {1}]
```

$\{\gamma_0 == (1+\theta^2)\,\sigma_\epsilon^2,\ \gamma_1 == \theta\,\sigma_\epsilon^2\}$

```
Reduce[γEquations, {θ, σ_ϵ^2}]
```

$$(\gamma_1 == 0 \;\&\&\; \gamma_0 == 0 \;\&\&\; \sigma_\epsilon^2 == 0) \;||\; \left(\gamma_1 \neq 0 \;\&\&\; \left(\theta == \frac{\gamma_0 - \sqrt{\gamma_0^2 - 4\gamma_1^2}}{2\gamma_1} \;||\; \theta == \frac{\gamma_0 + \sqrt{\gamma_0^2 - 4\gamma_1^2}}{2\gamma_1}\right) \;\&\&\; \sigma_\epsilon^2 == \gamma_0 - \gamma_1\,\theta\right) \;||\; (\gamma_1 == 0 \;\&\&\; \gamma_0 \neq 0 \;\&\&\; \theta == 0 \;\&\&\; \sigma_\epsilon^2 == \gamma_0)$$

```
s = Solve[γEquations, {θ, σ_ϵ^2}]; s // TableForm
```

$$\begin{array}{ll} \sigma_\epsilon^2 \to \frac{1}{2}\left(\gamma_0 - \sqrt{\gamma_0^2 - 4\gamma_1^2}\right) & \theta \to \frac{\gamma_0 + \sqrt{\gamma_0^2 - 4\gamma_1^2}}{2\gamma_1} \\ \sigma_\epsilon^2 \to \frac{\gamma_0}{2} + \frac{1}{2}\sqrt{\gamma_0^2 - 4\gamma_1^2} & \theta \to \frac{\gamma_0 - \sqrt{\gamma_0^2 - 4\gamma_1^2}}{2\gamma_1} \end{array}$$

With some hypothetical values, the MA parameters are

```
s /. {γ0 → 1, γ1 → -.2} // TableForm
```

$$\begin{array}{ll} \sigma_\epsilon^2 \to 0.0417424 & \theta \to -4.79129 \\ \sigma_\epsilon^2 \to 0.958258 & \theta \to -0.208712 \end{array}$$

So the second solution is the invertible one.

```
s // Simplify
```

$$\left\{\left\{\sigma_\epsilon^2 \to \frac{1}{2}\left(\gamma_0 - \sqrt{\gamma_0^2 - 4\gamma_1^2}\right),\; \theta \to \frac{\gamma_0 + \sqrt{\gamma_0^2 - 4\gamma_1^2}}{2\gamma_1}\right\},\; \left\{\sigma_\epsilon^2 \to \frac{1}{2}\left(\gamma_0 + \sqrt{\gamma_0^2 - 4\gamma_1^2}\right),\; \theta \to \frac{\gamma_0 - \sqrt{\gamma_0^2 - 4\gamma_1^2}}{2\gamma_1}\right\}\right\}$$

Alternative form for σ_ϵ^2 in the invertible case:

$$\text{Simplify}\left[\frac{1}{2}\left(\gamma_0 + \sqrt{\gamma_0^2 - 4\gamma_1^2}\right) == \frac{2\gamma_1^2}{\gamma_0 - \sqrt{\gamma_0^2 - 4\gamma_1^2}}\right]$$

```
True
```

Autoregressive forms

We define a recursive rule for the residuals:

$$\epsilon_{Rule} = \epsilon_{t\_} :\to \Delta p_t - \theta\,\epsilon_{t-1};$$

$\Delta p_t = \epsilon_t - \theta\epsilon_{t-1}$. With one recursive substitution for ϵ_{t-1}....

```
ϵt + (θ ϵt-1 /. ϵRule) // Expand
```

$\theta\,\Delta p_{-1+t} - \theta^2\,\epsilon_{-2+t} + \epsilon_t$

... and two recursive substitutions ...

```
ϵt + (θ ϵt-1 /. ϵRule /. ϵRule) // Expand
```

$-\theta^2\,\Delta p_{-2+t} + \theta\,\Delta p_{-1+t} + \theta^3\,\epsilon_{-3+t} + \epsilon_t$

Alternatively, going directly to the autoregressive form:

```
Series[(1 + θ L)^-1, {L, 0, 4}]
```

$1 - \theta L + \theta^2 L^2 - \theta^3 L^3 + \theta^4 L^4 + O[L]^5$

Exercise 4.1

Here is a function to build a table of all possible n successive realizations of q.

```
qTable[n_] := Table[(-1)^IntegerDigits[i, 2, n], {i, 0, 2^n - 1}];
```

... and for the 3-period problem, the realizations are:

```
q = qTable[3];
TableForm[q, TableHeadings → {Automatic, {"q0", "q1", "q2"}},
 TableAlignments → Right]
```

	q_0	q_1	q_2
1	1	1	1
2	1	1	-1
3	1	-1	1
4	1	-1	-1
5	-1	1	1
6	-1	1	-1
7	-1	-1	1
8	-1	-1	-1

The transition probabilities for each path are:

```
PrTrans[q_] := Table[If[q[[i, j]] == q[[i, j - 1]], α, 1 - α],
   {i, Length[q]}, {j, 2, Dimensions[q][[2]]}];
qp = PrTrans[q];
TableForm[qp, TableHeadings → {Automatic, {"Pr0→1", "Pr1→2"}},
 TableAlignments → Right]
```

	$Pr_{0\to1}$	$Pr_{1\to2}$
1	α	α
2	α	$1-\alpha$
3	$1-\alpha$	$1-\alpha$
4	$1-\alpha$	α
5	$1-\alpha$	α
6	$1-\alpha$	$1-\alpha$
7	α	$1-\alpha$
8	α	α

The total probabilities of each path are:

```
TotalProbs = Apply[Times, qp, {1}] / 2;
TableForm[TotalProbs,
 TableHeadings → {Automatic, {"PrTotal"}}, TableAlignments → Right]
```

1	$\frac{\alpha^2}{2}$
2	$\frac{1}{2}(1-\alpha)\,\alpha$
3	$\frac{1}{2}(1-\alpha)^2$
4	$\frac{1}{2}(1-\alpha)\,\alpha$
5	$\frac{1}{2}(1-\alpha)\,\alpha$
6	$\frac{1}{2}(1-\alpha)^2$
7	$\frac{1}{2}(1-\alpha)\,\alpha$
8	$\frac{\alpha^2}{2}$

Verify that the probabilities sum to one:

```
Total[TotalProbs] // Simplify
```

```
1
```

Compute the v_t's using the definition $v_t = q_t - \phi q_{t-1}$:

```
v = q[[All, {2, 3}]] - ϕ q[[All, {1, 2}]];
TableForm[v, TableHeadings → {Automatic, {"v1", "v2"}}, TableAlignments → Right]
```

	v_1	v_2
1	$1-\phi$	$1-\phi$
2	$1-\phi$	$-1-\phi$
3	$-1-\phi$	$1+\phi$
4	$-1-\phi$	$-1+\phi$
5	$1+\phi$	$1-\phi$
6	$1+\phi$	$-1-\phi$
7	$-1+\phi$	$1+\phi$
8	$-1+\phi$	$-1+\phi$

Verify that $Ev_1 = Ev_2 = 0$:

```
TotalProbs.v[[All]] // Simplify
```

{0, 0}

$\text{Var}(v_t) \equiv \gamma_0$

```
γ0 = (TotalProbs.(v[[All]]^2) // Simplify)
```

$\{1 + (2 - 4\alpha)\phi + \phi^2, 1 + (2 - 4\alpha)\phi + \phi^2\}$

$\text{Cov}(v_t, v_{t-1}) = E v_t v_{t-1} \equiv \gamma_1$

```
γ1 = (TotalProbs.(v[[All, 1]] v[[All, 2]]) // Simplify)
```

$-4\alpha^2\phi - (1+\phi)^2 + 2\alpha(1+\phi)^2$

The v_t must be uncorrelated, so solve for the value of ϕ that makes $\gamma_1 = 0$:

```
s = Solve[γ1 == 0, ϕ]
```

$\left\{\left\{\phi \to \frac{1}{-1+2\alpha}\right\}, \{\phi \to -1+2\alpha\}\right\}$

There are two solutions, but only the second has $|\phi| < 1$. For example,

```
ϕ /. s /. α → .6
```

{5., 0.2}

So, take the second solution:

```
s = s[[2, 1]]
```

$\phi \to -1 + 2\alpha$

Verify that this results in $\gamma_1 = 0$:

```
TotalProbs.(v[[All, 1]] v[[All, 2]]) /. s // Simplify
```

```
0
```

The v_t's have no skewness:

```
TotalProbs.(v[[All, 2]]^3) /. s // Simplify
```

```
0
```

Verify that the higher-order serial moment is non-zero, i.e., $\text{Cov}\left(v_{t-1},\ v_t^3\right) = E v_{t-1}\, v_t^3 \neq 0$

```
TotalProbs.(v[[All, 1]] * v[[All, 2]]^3) /. s // Simplify
```

$$-32\ (-1+\alpha)^2\ \alpha^2\ (-1+2\ \alpha)$$

```
ClearAll[q, u, v, γ0, γ1]
```

Exercise 4.2

The model is:

```
mRule = m_t_ :→ m_{t-1} + u_t;
pRule = p_t_ :→ m_t + c q_t;
ΔpRule = Δp_t_ :→ (p_t /. pRule /. mRule) - (p_{t-1} /. pRule);
Δp_t /. ΔpRule
```

$$-c\ q_{-1+t} + c\ q_t + u_t$$

We need some alternate rules for the expectation operator to recognize the correlation between q_t and q_{t-1}.

```
ℰAlternateRules = {
   ℰ[q_^2] → 1, ℰ[u_^2] → σu^2,
   ℰ[q_ u_] → 0, ℰ[q_t_ q_s_] :→ 0 /; Abs[t - s] > 1,
   ℰ[q_t_ q_s_] :→ ρ /; Abs[t - s] == 1,
   ℰ[u_t_ u_s_] :→ 0 /; t =!= s};
```

```
ℰRules = Join[ℰAlternateRules, ℰLinearityRules];
```

To get the variance, we multiply everything out, and take the expectation:

```
ℰ[Expand[Δp_t^2 /. ΔpRule]]
```

$$\mathcal{E}\left[c^2\ q_{-1+t}^2 - 2\ c^2\ q_{-1+t}\ q_t + c^2\ q_t^2 - 2\ c\ q_{-1+t}\ u_t + 2\ c\ q_t\ u_t + u_t^2\right]$$

Using the rules described about to eliminate terms that have zero expectation:

```
% //. εRules // Simplify
```

$-2\,c^2\,(-1+\rho)+\sigma_u^2$

Cov $(\Delta p_t,\ \Delta p_{t-1})$:

```
ε[Expand[Δp_t Δp_{t-1} /. ΔpRule]]
```

$\mathcal{E}[c^2\,q_{-2+t}\,q_{-1+t}-c^2\,q_{-1+t}^2-c^2\,q_{-2+t}\,q_t+c^2\,q_{-1+t}\,q_t-c\,q_{-1+t}\,u_{-1+t}+c\,q_t\,u_{-1+t}-c\,q_{-2+t}\,u_t+c\,q_{-1+t}\,u_t+u_{-1+t}\,u_t]$

```
ε[Expand[Δp_t Δp_{t-1} /. ΔpRule]] //. εRules // FullSimplify
```

$c^2\,(-1+2\,\rho)$

```
ε[Expand[Δp_t Δp_{t-2} /. ΔpRule]] //. εRules // Simplify
```

$-c^2\,\rho$

```
ε[Expand[Δp_t Δp_{t-3} /. ΔpRule]] //. εRules // Simplify
```

0

Verify that $\sqrt{-\mathrm{Cov}\,(\Delta p_t,\ \Delta p_{t-1})}<c$:

```
Simplify[Sqrt[-ε[Expand[Δp_t Δp_{t-1} /. ΔpRule]] //. εRules] < c,
 Assumptions → {0 < ρ < 1/2, c > 0}]
```

True

Exercise 4.3

Model:

```
mRule = m_t_ :→ m_{t-1} + u_t;
pRule = p_t_ :→ m_t + c q_t;
ΔpRule = Δp_t_ :→ (p_t /. pRule /. mRule) - (p_{t-1} /. pRule);
Δp_t /. ΔpRule
```

$-c\,q_{-1+t}+c\,q_t+u_t$

Modified expectations rules:

```
ℰAlternate_Rules = {
   ℰ[q_^2] → 1,
   ℰ[u_^2] → σ_u^2,
   ℰ[q_s_ u_t_] :→ ρ σ_u /; t == s,
   ℰ[q_s_ u_t_] :→ 0 /; t =!= s,
   ℰ[q_t_ q_s_] :→ 0 /; t =!= s,
   ℰ[u_t_ u_s_] :→ 0 /; t =!= s};
```

```
ℰ_Rules = Join[ℰAlternate_Rules, ℰLinearity_Rules];
```

```
ℰ[Expand[Δp_t^2 /. Δp_Rule]] //. ℰ_Rules // FullSimplify
```

$2\,c^2 + \sigma_u^2 + 2\,c\,\rho\,\sigma_u$

```
ℰ[Expand[Δp_t Δp_{t-1} /. Δp_Rule]] //. ℰ_Rules // Simplify
```

$-c\,(c + \rho\,\sigma_u)$

```
ℰ[Expand[Δp_t Δp_{t-2} /. Δp_Rule]] //. ℰ_Rules // Simplify
```

0

Verify that $\sqrt{-\mathrm{Cov}\,(\Delta p_t,\,\Delta p_{t-1})} > c$:

```
Simplify[√(-ℰ[Expand[Δp_t Δp_{t-1} /. Δp_Rule]] //. ℰ_Rules) > c,
  Assumptions → {0 < ρ, σ_u > 0, c > 0}]
```

True

■ Generalized Roll Model (Chapter 8)

Assign 'constant' property to model parameters.

```
Attributes[c] = {Constant};
Attributes[λ] = {Constant};
```

Structural model

```
mRule = mt_ :> mt-1 + wt; wRule = wt_ :> λ qt + ut; pRule = pt_ :> mt + c qt;
TableForm @ {mRule, wRule, pRule}
```

$m_{t\_} \mapsto m_{t-1} + w_t$
$w_{t\_} \mapsto \lambda\, q_t + u_t$
$p_{t\_} \mapsto m_t + c\, q_t$

The bid and ask are:

```
pt /. pRule /. mRule /. wRule /. qt → {-1, 1} // TableForm
```

$-c - \lambda + m_{-1+t} + u_t$
$c + \lambda + m_{-1+t} + u_t$

(The spread is $2\,(c + \lambda)$.) The price change:

```
ΔpRule = Δpt_ :> (pt /. pRule /. mRule /. wRule) - (pt-1 /. pRule);
Δpt /. ΔpRule
```

$-c\, q_{-1+t} + c\, q_t + \lambda\, q_t + u_t$

Autocovariances: $\mathrm{Var}[\Delta p_t] = \gamma_0$:

```
ℰ[Expand[Δpt^2 /. ΔpRule]]
```

$\mathcal{E}[c^2 q_{-1+t}^2 - 2 c^2 q_{-1+t} q_t - 2 c \lambda q_{-1+t} q_t + c^2 q_t^2 + 2 c \lambda q_t^2 + \lambda^2 q_t^2 - 2 c\, q_{-1+t} u_t + 2 c\, q_t u_t + 2 \lambda q_t u_t + u_t^2]$

```
ℰRules = Join[ℰRollRules, ℰLinearityRules];
```

```
ℰ[Expand[Δpt^2 /. ΔpRule]] //. ℰRules
```

$2 c^2 + 2 c \lambda + \lambda^2 + \sigma_u^2$

$\mathrm{Cov}[\Delta p_t, \Delta p_{t-1}] = \gamma_1$:

```
ℰ[Expand[Δpt Δpt-1 /. ΔpRule]] //. ℰRules
```

$-c^2 - c \lambda$

$\mathrm{Cov}[\Delta p_t, \Delta p_{t-2}] = \gamma_2$:

```
ℰ[Expand[Δpt Δpt-2 /. ΔpRule]] //. ℰRules
```

0

Summarize the first two autocovariances in terms of the structural parameters:

```
γStructural_Rules = {γ0 → 2 c^2 + 2 c λ + λ^2 + σu^2, γ1 → -c^2 - c λ};
γStructural_Rules // TableForm
```

$\gamma_0 \to 2\,c^2 + 2\,c\,\lambda + \lambda^2 + \sigma_u^2$
$\gamma_1 \to -c^2 - c\,\lambda$

Two γs don't suffice to identify c, λ, σ_u^2.

Special case. If λ=0 (all public info), this reduces to the original Roll model.

Special case. If $\sigma_u^2 = 0$, there is no public information; only private information.

Now evaluate $\mathrm{Var}[w_t] = \sigma_w^2$:

```
ℰ[Expand[w_t^2 /. w_Rule]] //. ℰ_Rules
```

$\lambda^2 + \sigma_u^2$

Compare with:

```
γ0 + 2 γ1 /. γStructural_Rules // Simplify
```

$\lambda^2 + \sigma_u^2$

(Variance ratios.)

Forecasting using structural and statistical representations

Analyze
$f_t \equiv \lim_{k\to\infty} \mathcal{E}^*[p_{t+k} \mid p_t, \ldots]$ (show) $= \mathcal{E}^*[m_t \mid p_t, \ldots] = p_t - c\,\mathcal{E}^*[q_t \mid p_t, \ldots]$

Analyze $\mathcal{E}^*[q_t \mid p_t, \ldots] = \left(\beta_0 + \beta_1 L + \beta_2 L^2 + \ldots\right) \epsilon_t$.

In the projection $\beta_k = \frac{\mathrm{Cov}(\epsilon_{t-k}, q_t)}{\sigma_\epsilon^2}$

To compute $\mathrm{Cov}(\epsilon_{t-k}, q_t)$, note that the statistical representation for Δp_t is

$\Delta p_t = \epsilon_t + \theta\epsilon_{t-1} = (1 + \theta L)\,\epsilon_t$;

the structural representation for Δp_t is $\Delta p_t = (c\,(1 - L) + \lambda)\,q_t + u_t$.

Equating the structural and statistical representations:
$(1 + \theta L)\,\epsilon_t = (c\,(1 - L) + \lambda)\,q_t + u_t$, or
$\epsilon_t = (1 - \theta L)^{-1}\,(c\,(1 - L) + \lambda)\,q_t + (1 - \theta L)^{-1}\,u_t$

That is, $\epsilon_t = A(L)\,q_t + B(L)\,u_t$ where (expanding through the fifth term) $A(L) =$

```
Series[(1 + θ L)^-1 (c (1 - L) + λ), {L, 0, 5}] // Simplify
```

$(c+\lambda)+(-c-c\,\theta-\theta\,\lambda)\,L+\theta\,(c+c\,\theta+\theta\,\lambda)\,L^2-\theta^2\,(c+c\,\theta+\theta\,\lambda)\,L^3+\theta^3\,(c+c\,\theta+\theta\,\lambda)\,L^4-\theta^4\,(c+c\,\theta+\theta\,\lambda)\,L^5+O[L]^6$

i.e., $\mathrm{Cov}\,(\epsilon_{t-k},\ q_t) = \mathrm{Cov}\,(A_0\,q_{t-k}+A_1\,q_{t-k-1}+...+B\,(L)\,u_t,\ q_t) = \begin{cases} c+\lambda, & \text{if } k=0 \\ 0, & \text{otherwise}\end{cases}$

Therefore, $\beta_0 = \frac{c+\lambda}{\sigma_\epsilon^2}$ and $\beta_k = 0\ \forall\ k > 0$ and $f_t = p_t - \frac{c\,(c+\lambda)}{\sigma_\epsilon^2}\ \epsilon_t$

Does this equal the same forecast we'd compute from the statistical representation, $f_t = p_t + \theta\epsilon_t$? Determine θ in terms of the structural parameters.

The autocovariances computed from the statistical and structural representations must agree. The autocovariances for the MA(1) process $\triangle p_t = \epsilon_t + \theta\epsilon_{t-1}$ are:

```
γStatistical_Rules = {γ0 → (θ^2 + 1) σϵ^2, γ1 → θ σϵ^2}; γStatistical_Rules // TableForm
```

$\gamma_0 \to (1+\theta^2)\ \sigma_\epsilon^2$

$\gamma_1 \to \theta\ \sigma_\epsilon^2$

```
StatStructEqu = Apply[Equal, Join[γStatistical_Rules, γStructural_Rules], {1}];
StatStructEqu // TableForm
```

$\gamma_0 == (1+\theta^2)\ \sigma_\epsilon^2$

$\gamma_1 == \theta\ \sigma_\epsilon^2$

$\gamma_0 == 2\,c^2 + 2\,c\,\lambda + \lambda^2 + \sigma_u^2$

$\gamma_1 == -c^2 - c\,\lambda$

which implies:

```
Solve[Eliminate[StatStructEqu, {σu^2, γ0, γ1}], θ]
```

$\left\{\left\{\theta \to -\frac{c\,(c+\lambda)}{\sigma_\epsilon^2}\right\}\right\}$

i.e., the coefficient of ϵ_t is the same in both representations.
Special case: If $\sigma_u^2 = 0$ (no public information) $f_t = m_t$.

The pricing error: How closely does p_t track m_t?

The structural model implies $s_t = q_t\,c$, so $\sigma_s^2 = c^2$, but neither c nor σ_u^2 are identified.

Lower bound

$s_t = p_t - m_t = (p_t - f_t) - (m_t - f_t)$. Since f_t is a linear projection of m_t on $\{p_t, p_{t-1}, \ldots\}$, the filtering error $m_t - f_t$ is uncorrelated with $p_t - f_t$.
Therefore $\sigma_s^2 = \mathrm{Var}(p_t - f_t) + \mathrm{Var}(m_t - f_t)$.

Next we use the property that $f_t = p_t + \theta \epsilon_t$ is not dependent on the structural model parameters.
$\Rightarrow \mathrm{Var}(p_t - f_t) = \theta^2 \sigma_\epsilon^2$ is invariant to structural identificatation
Under one parameterization (that of exclusively private information, $u_t = 0$), $m_t - f_t = 0$.
This parameterization defines the lower bound.
If $u_t = 0$, we've seen that $m_t = f_t = p_t + \theta \epsilon_t$, so $\sigma_s^2 = \theta^2 \sigma_\epsilon^2 = c^2$. To establish the last equality, recall that we have a mapping from the structural parameters to the autocovariances, and from the autocovariances to the moving average parameters.
The MA parameters in terms of the structural parameters are:

```
sol = Solve[StatStructEqu, {θ, σϵ²}, {γ0, γ1}] // Simplify
```

$$\left\{\left\{\sigma_\epsilon^2 \to \frac{1}{2}\left(2c^2 + 2c\lambda + \lambda^2 + \sigma_u^2 - \sqrt{\lambda^2 + \sigma_u^2}\sqrt{4c^2 + 4c\lambda + \lambda^2 + \sigma_u^2}\right), \theta \to -\frac{2c^2 + 2c\lambda + \lambda^2 + \sigma_u^2 + \sqrt{\lambda^2 + \sigma_u^2}\sqrt{4c^2 + 4c\lambda + \lambda^2 + \sigma_u^2}}{2c(c+\lambda)}\right\}, \left\{\sigma_\epsilon^2 \to \frac{1}{2}\left(2c^2 + 2c\lambda + \lambda^2 + \sigma_u^2 + \sqrt{\lambda^2 + \sigma_u^2}\sqrt{4c^2 + 4c\lambda + \lambda^2 + \sigma_u^2}\right), \theta \to -\frac{2c^2 + 2c\lambda + \lambda^2 + \sigma_u^2 - \sqrt{\lambda^2 + \sigma_u^2}\sqrt{4c^2 + 4c\lambda + \lambda^2 + \sigma_u^2}}{2c(c+\lambda)}\right\}\right\}$$

```
InvertibleSolution = sol[[2]];
```

The value computed for the lower bound is $\theta^2 \sigma_\epsilon^2$:

```
FullSimplify[θ² σϵ² /. InvertibleSolution, {σu² > 0, c > 0, λ > 0}]
```

$$\frac{1}{2}\left(2c^2 + 2c\lambda + \lambda^2 + \sigma_u^2 - \sqrt{(\lambda^2 + \sigma_u^2)((2c+\lambda)^2 + \sigma_u^2)}\right)$$

In the case of exclusively private information, the lower bound is correct.

```
FullSimplify[θ² σϵ² /. InvertibleSolution, {σu² == 0, c > 0, λ > 0}]
```

$$c^2$$

In the case of exclusively public information $(\sigma_u^2 \neq 0, \lambda = 0)$, though, the lower bound is (in terms of the structural parameters):

```
FullSimplify[θ^2 σϵ^2 /. InvertibleSolution, {σu^2 > 0, c > 0, λ == 0}]
```

$$\frac{1}{2}\left(2\,c^2+\sigma_u^2-\sqrt{\sigma_u^2\left(4\,c^2+\sigma_u^2\right)}\right)$$

Verify that this is less than the true $\sigma_s^2 = c^2$:

```
Simplify[% < c^2, {σu^2 > 0, c > 0, λ == 0}]
```

```
True
```

General univariate random-walk decompositions (Section 8.6)

```
Clear[s, sol, K];(* Remove variables that may have been used earlier. *)
```

Moving average operator

```
θRules = {θ[z_, q_: q, j_: j] :> Sum[θj z^j, {j, 0, q}] /; FreeQ[z, Subscript],
    θ[z_t_, q_: q, j_: j] :> Sum[θj z_{t-j}, {j, 0, q}]};
```

This may be used to define an MA polynomial in the lag operator

```
θ[L, 5] /. θRules
```

$$\theta_0 + L\,\theta_1 + L^2\,\theta_2 + L^3\,\theta_3 + L^4\,\theta_4 + L^5\,\theta_5$$

or to generate a moving average

```
θ[ϵt] /. θRules
```

$$\sum_{j=0}^{q} \epsilon_{-j+t}\,\theta_j$$

Define a lag operator L that applies only to time series x:

```
LAppliesTo[x_] := {L^k_ x_t_ :> x_{t-k}, L x_t_ :> x_{t-1}}
```

Note that *Mathematica* thinks that L^k is a quantity in a multiplication, and reorders in its standard fashion:

```
(θ[L, 5] /. θRules) ϵt
```

$$\epsilon_t\,\left(\theta_0 + L\,\theta_1 + L^2\,\theta_2 + L^3\,\theta_3 + L^4\,\theta_4 + L^5\,\theta_5\right)$$

```
Expand[%]
```

$\epsilon_t\,\theta_0 + L\,\epsilon_t\,\theta_1 + L^2\,\epsilon_t\,\theta_2 + L^3\,\epsilon_t\,\theta_3 + L^4\,\epsilon_t\,\theta_4 + L^5\,\epsilon_t\,\theta_5$

To invoke the L operator:

```
% /. LAppliesTo[ϵ]
```

$\epsilon_t\,\theta_0 + \epsilon_{-1+t}\,\theta_1 + \epsilon_{-2+t}\,\theta_2 + \epsilon_{-3+t}\,\theta_3 + \epsilon_{-4+t}\,\theta_4 + \epsilon_{-5+t}\,\theta_5$

Setup

The structural model is:

$p_t = m_t + s_t$

$m_t = m_{t-1} + u_t$

or to generate a moving average

The statistical representation for Δp_t is a moving average of order *K*:

$\Delta p_{\text{Rule}} = \Delta p_{t\_} \mapsto \theta[\epsilon_t, K];$

Using this rule, and the rule for the moving average operator Δp_t =

Δp_t /. Δp_{Rule} /. θ_{Rules}

$$\sum_{j=0}^{K} \epsilon_{-j+t}\,\theta_j$$

Forecasting

The price change *k* periods ahead of *s* is:

Δp_{s+k} /. Δp_{Rule} /. θ_{Rules}

$$\sum_{j=0}^{K} \epsilon_{-j+k+s}\,\theta_j$$

The price at time p_{t+K} is p_s plus the cumulative change through $s + K$:

$$p_s + \sum_{k=1}^{K} \left(\Delta p_{s+k} \text{ /. } \Delta p_{\text{Rule}} \text{ /. } \theta_{\text{Rules}}\right)$$

$$p_s + \sum_{k=1}^{K} \sum_{j=0}^{K} \epsilon_{-j+k+s}\,\theta_j$$

For example, let

```
K = 3;
```

```
p_s + Sum[(Δp_{s+k} /. Δp_Rule /. θ_Rules), {k, 1, K}]
```

$p_s + \epsilon_{1+s}\theta_0 + \epsilon_{2+s}\theta_0 + \epsilon_{3+s}\theta_0 + \epsilon_s\theta_1 + \epsilon_{1+s}\theta_1 + \epsilon_{2+s}\theta_1 + \epsilon_{-1+s}\theta_2 + \epsilon_s\theta_2 + \epsilon_{1+s}\theta_2 + \epsilon_{-2+s}\theta_3 + \epsilon_{-1+s}\theta_3 + \epsilon_s\theta_3$

...grouping by ϵ's:

```
Collect[%, Table[ϵ_k, {k, s - K - 1, s + K}]]
```

$p_s + \epsilon_{3+s}\theta_0 + \epsilon_{2+s}(\theta_0 + \theta_1) + \epsilon_{1+s}(\theta_0 + \theta_1 + \theta_2) + \epsilon_{-2+s}\theta_3 + \epsilon_{-1+s}(\theta_2 + \theta_3) + \epsilon_s(\theta_1 + \theta_2 + \theta_3)$

To construct a forecast (as of time s), use the fact that $\mathcal{E}[\epsilon_k \mid p_s, p_{s-1}, \ldots] = 0 \ \forall\ k > s$

```
% /. ϵ_k_ :> 0 /; Simplify[k > s]
```

$p_s + \epsilon_{-2+s}\theta_3 + \epsilon_{-1+s}(\theta_2 + \theta_3) + \epsilon_s(\theta_1 + \theta_2 + \theta_3)$

This can all be summarized in the forecast function

```
f_Rule = f_s_ :> (Collect[p_s + Sum[(Δp_{s+k} /. Δp_Rule /. θ_Rules), {k, 1, K}],
      Table[ϵ_k, {k, s - K - 1, s + K}]] /. ϵ_k_ :> 0 /; Simplify[k > s]);
```

```
f_t /. f_Rule
```

$p_t + \epsilon_{-2+t}\theta_3 + \epsilon_{-1+t}(\theta_2 + \theta_3) + \epsilon_t(\theta_1 + \theta_2 + \theta_3)$

The first difference of the forecast is proportional to ϵ_t, and so is a M'gale:

```
f_t - f_{t-1} /. f_Rule /. (p_t - p_{t-1}) → (Δp_t /. Δp_Rule /. θ_Rules) // Simplify
```

$\epsilon_t(\theta_0 + \theta_1 + \theta_2 + \theta_3)$

when $m_t = f_t$, the above is $= w_t$ and $\sigma_w^2 =$

```
Coefficient[%, ϵ_t]^2 σ_ϵ^2
```

$\sigma_\epsilon^2(\theta_0 + \theta_1 + \theta_2 + \theta_3)^2$

cf. eq. (8.10) in the text.

□ *Pricing error (when $m_t = f_t$)*

```
p_t - m_t /. m_t → f_t /. f_Rule
```

$-\epsilon_{-2+t}\theta_3 - \epsilon_{-1+t}(\theta_2 + \theta_3) - \epsilon_t(\theta_1 + \theta_2 + \theta_3)$

The lower-bound for σ_s^2 is:

```
Plus @@ Table[Coefficient[%, ϵ_{t-i}], {i, 0, K - 1}]^2 σ_ϵ^2
```

$\sigma_\epsilon^2 \left((-\theta_2 - \theta_3)^2 + (-\theta_1 - \theta_2 - \theta_3)^2 + \theta_3^2\right)$

cf. (8.12) in the text.

Exercise 8.1

The model is observationally equivalent to one in which there is no lag on the efficient price. The autocovariances and moving average representation are the same.

Exercise 8.2

By rearranging, the model can be written as $(1 - (1 - \alpha) L) p_t = \alpha m_t$. Taking first differences $(1 - (1 - \alpha) L) \Delta p_t = \epsilon_t = \alpha w_t$ So:

```
ϕ[L_] := 1 - (1 - α) L
```

The MA representation is $\Delta p_t = \theta(L) \epsilon_t$ where $\theta(L) = \phi(L)^{-1}$. Furthermore $\theta(1)^2 =$

```
ϕ[1]^-2
```

$\frac{1}{\alpha^2}$

Since $\sigma_\epsilon^2 = \alpha^2 \sigma_w^2$, $\theta(1)^2 \sigma_\epsilon^2 = \sigma_w^2$

Exercise 8.3

Over five-minute intervals

```
σRule = σ_ϵ^2 → 0.00001;
```

```
θ[L_] := 1 - 0.3 L + 0.1 L^2
```

Random-walk variance:

```
Sqrt[θ[1]^2 σ_ϵ^2 /. σRule]
```

```
0.00252982
```

Over one day:

```
Sqrt[6 * 12 * θ[1]^2 σϵ^2 /. σRule]
```

```
0.0214663
```

i.e., about 2%

For the pricing error variance, the C_i coefficients are generally:

```
CRule = C[θ_, i_] :> Sum[-Coefficient[θ[L], L, j], {j, i+1, Exponent[θ[L], L]}];
```

and here ...

```
Table[C[θ, i], {i, 0, 2}] /. CRule
```

```
{0.2, -0.1, 0}
```

```
Sqrt[Sum[C[θ, i]^2 σϵ^2, {i, 0, Exponent[θ[L], L] - 1}] /. CRule /. σRule]
```

```
0.000707107
```

i.e., about seven basis points

Exercise 8.4

The structural model is:

$m_t = m_{t-1} + w_t$

$w_t = \lambda q_t + u_t$

$p_t = m_{t-1} + c\, q_t$

Notice that the price is determined with respect to *lagged* value of the implicit efficient price.

(a) Using the structural representation, determine the Δp_t autocovariances $\gamma_0,\ \gamma_1$ and verify that $\gamma_2 = 0$.

In this and the following parts, assume that $c = 2$ and $\lambda = 1$.

(b) Verify that autocovariances are the same as the autocovariances for the (statistical) MA(1) model $\Delta p_t = \epsilon_t + \Theta$ where

$$\sigma_\epsilon^2 = \frac{1}{2}\left(\sigma_u^2 + \sqrt{\left(\sigma_u^2+1\right)\left(\sigma_u^2+9\right)} + 5\right)$$

and

$$\theta = \frac{1}{4}\left(-\sigma_u^2 + \sqrt{\left(\sigma_u^2+1\right)\left(\sigma_u^2+9\right)} - 5\right)$$

(c) Verify that $\sigma_w^2 = (1+\Theta)^2\, \sigma_\epsilon^2$.

(d) Compute (in terms of the MA parameters) the lower bound for σ_s^2 where $s_t = p_t - m_t$. Verify that the lower bound is exact when $\sigma_u^2 = 0$.

Analysis

```
nValues = {c → 2, λ → 1};
```

```
mRule = m_t_ :→ m_(t-1) + w_t;
wRule = w_t_ :→ λ q_t + u_t;
pRule = p_t_ :→ m_(t-1) + c q_t;
```

Pricing error s_t =

```
p_t - m_t /. pRule /. m_t → (m_t /. mRule) /. wRule
```

$c\, q_t - \lambda\, q_t - u_t$

This implies that the pricing error variance is

```
(c - λ)^2 + σ_u^2 /. nValues
```

$1 + \sigma_u^2$

Price changes:

```
ΔpRule = Δp_t_ :→ (p_t /. pRule /. mRule /. wRule) - (p_(t-1) /. pRule);
Δp_t /. ΔpRule
```

$-c\, q_{-1+t} + \lambda\, q_{-1+t} + c\, q_t + u_{-1+t}$

$\mathrm{Var}[\Delta p_t] = \gamma_0$:

```
ℰ[Expand[Δp_t^2 /. ΔpRule]] //. ℰRules
```

$2\, c^2 - 2\, c\, \lambda + \lambda^2 + \sigma_u^2$

```
2 c^2 - 2 λ c + λ^2 + σ_u^2
```

$2\, c^2 - 2\, c\, \lambda + \lambda^2 + \sigma_u^2$

$\mathrm{Cov}[\Delta p_t, \Delta p_{t-1}] = \gamma_1$:

```
ℰ[Expand[Δp_t Δp_(t-1) /. ΔpRule]] //. ℰRules
```

$-c^2 + c\, \lambda$

$\mathrm{Cov}[\Delta p_t, \Delta p_{t-2}] = \gamma_2$:

```
ℰ[Expand[Δp_t Δp_{t-2} /. Δp_Rule]] //. ℰ_Rules
```

0

Summarize the first two autocovariances in terms of the structural parameters:

```
γStructural_Rules = {γ_0 → 2 c^2 - 2 λ c + λ^2 + σ_u^2, γ_1 → -c^2 + c λ};
γStructural_Rules // TableForm
```

$\gamma_0 \to 2\,c^2 - 2\,c\,\lambda + \lambda^2 + \sigma_u^2$

$\gamma_1 \to -c^2 + c\,\lambda$

Now evaluate $\mathrm{Var}[w_t] = \sigma_w^2$:

```
ℰ[Expand[w_t^2 /. w_Rule]] //. ℰ_Rules
```

$\lambda^2 + \sigma_u^2$

The autocovariances computed from the statistical and structural representations must agree. The autocovariances for the MA(1) process $\Delta p_t = \epsilon_t + \theta \epsilon_{t-1}$ are:

```
γStatistical_Rules = {γ_0 → (θ^2 + 1) σ_ϵ^2, γ_1 → θ σ_ϵ^2}; γStatistical_Rules // TableForm
```

$\gamma_0 \to (1 + \theta^2)\,\sigma_\epsilon^2$

$\gamma_1 \to \theta\,\sigma_\epsilon^2$

```
StatStructEqu = Apply[Equal, Join[γStatistical_Rules, γStructural_Rules], {1}];
StatStructEqu // TableForm
```

$\gamma_0 == (1 + \theta^2)\,\sigma_\epsilon^2$

$\gamma_1 == \theta\,\sigma_\epsilon^2$

$\gamma_0 == 2\,c^2 - 2\,c\,\lambda + \lambda^2 + \sigma_u^2$

$\gamma_1 == -c^2 + c\,\lambda$

which implies:

```
sol = Solve[StatStructEqu, {θ, σϵ^2}, {γ0, γ1}] // Simplify
```

$$\left\{\left\{\sigma_\epsilon^2 \to \frac{1}{2}\left(2\,c^2 - 2\,c\,\lambda + \lambda^2 + \sigma_u^2 - \sqrt{\lambda^2+\sigma_u^2}\,\sqrt{4\,c^2 - 4\,c\,\lambda + \lambda^2 + \sigma_u^2}\right),\right.\right.$$
$$\left.\theta \to -\frac{2\,c^2 - 2\,c\,\lambda + \lambda^2 + \sigma_u^2 + \sqrt{\lambda^2+\sigma_u^2}\,\sqrt{4\,c^2 - 4\,c\,\lambda + \lambda^2 + \sigma_u^2}}{2\,c^2 - 2\,c\,\lambda}\right\},$$
$$\left\{\sigma_\epsilon^2 \to \frac{1}{2}\left(2\,c^2 - 2\,c\,\lambda + \lambda^2 + \sigma_u^2 + \sqrt{\lambda^2+\sigma_u^2}\,\sqrt{4\,c^2 - 4\,c\,\lambda + \lambda^2 + \sigma_u^2}\right),\right.$$
$$\left.\left.\theta \to -\frac{2\,c^2 - 2\,c\,\lambda + \lambda^2 + \sigma_u^2 - \sqrt{\lambda^2+\sigma_u^2}\,\sqrt{4\,c^2 - 4\,c\,\lambda + \lambda^2 + \sigma_u^2}}{2\,c^2 - 2\,c\,\lambda}\right\}\right\}$$

```
InvertibleSolution = sol[[2]]
```

$$\left\{\sigma_\epsilon^2 \to \frac{1}{2}\left(2\,c^2 - 2\,c\,\lambda + \lambda^2 + \sigma_u^2 + \sqrt{\lambda^2+\sigma_u^2}\,\sqrt{4\,c^2 - 4\,c\,\lambda + \lambda^2 + \sigma_u^2}\right),\right.$$
$$\left.\theta \to -\frac{2\,c^2 - 2\,c\,\lambda + \lambda^2 + \sigma_u^2 - \sqrt{\lambda^2+\sigma_u^2}\,\sqrt{4\,c^2 - 4\,c\,\lambda + \lambda^2 + \sigma_u^2}}{2\,c^2 - 2\,c\,\lambda}\right\}$$

```
FullSimplify[InvertibleSolution /. nValues, {λ > 0, σu^2 > 0, c > 0}]
```

$$\left\{\sigma_\epsilon^2 \to \frac{1}{2}\left(5 + \sigma_u^2 + \sqrt{\left(1+\sigma_u^2\right)\left(9+\sigma_u^2\right)}\right),\ \theta \to \frac{1}{4}\left(-5 - \sigma_u^2 + \sqrt{\left(1+\sigma_u^2\right)\left(9+\sigma_u^2\right)}\right)\right\}$$

```
Simplify[ (1 + θ)^2 σϵ^2 /. InvertibleSolution]
```

$$\lambda^2 + \sigma_u^2$$

i.e., the coefficient of ϵ_t is the same in both representations.

```
Simplify[θ^2 σϵ^2 /. InvertibleSolution /. nValues, {λ > 0, σu^2 > 0, c > 0}]
```

$$\frac{1}{2}\left(5 + \sigma_u^2 - \sqrt{\left(1+\sigma_u^2\right)\left(9+\sigma_u^2\right)}\right)$$

```
Simplify[θ^2 σϵ^2 /. InvertibleSolution /. nValues /. σu^2 → 0, {λ > 0, σu^2 > 0, c > 0}]
```

$$1$$

```
Solve[(θ^2 σϵ^2 /. InvertibleSolution) == (c - λ)^2 + σu^2, σu^2]
```

$$\left\{\left\{\sigma_u^2 \to 0\right\}\right\}$$

```
InvertibleSolution /. nValues /. σ_u^2 → 0 // Simplify
```

$\left\{\sigma_{\epsilon}^{2} \to 4,\ \theta \to -\frac{1}{2}\right\}$