MultivariateMicrostructureModels

Empirical Market Microstructure

(2006, Oxford University Press)

Companion *Mathematica* notebook

Joel Hasbrouck

This notebook covers material in Chapters 9 and 10

```
Text @ Style["Notebook evaluated " <> DateString["DateTime"], "Subtitle"]
```

Notebook evaluated Monday 4 June 2007 20:28:52

Preliminaries

Notations

```
<< Notation`
```

The following commands define symbolizations that are convenient for labeling things.

```
Symbolize[Anything_Rule]; Symbolize[Anything_Rules];
Symbolize[Anything_Solution]; Symbolize[Anything_Solutions];
```

```
Symbolize[σ²_]
```

Additional initializations specific to this notebook

```
SetAttributes[c, Constant]
```

Permutation routine

The following routine is useful for permuting the order of variables in coefficient and covariance matrices.

```
Permute[m_, i_] := Transpose[Transpose[m[[i]]][[i]]]
```

For example:

```
(vTest = Table[σ_{Min[i,j],Max[i,j]}, {i, 4}, {j, 4}]) // MatrixForm
```

$$\begin{pmatrix} \sigma_{1,1} & \sigma_{1,2} & \sigma_{1,3} & \sigma_{1,4} \\ \sigma_{1,2} & \sigma_{2,2} & \sigma_{2,3} & \sigma_{2,4} \\ \sigma_{1,3} & \sigma_{2,3} & \sigma_{3,3} & \sigma_{3,4} \\ \sigma_{1,4} & \sigma_{2,4} & \sigma_{3,4} & \sigma_{4,4} \end{pmatrix}$$

```
Permute[vTest, {2, 4, 1, 3}] // MatrixForm
```

$$\begin{pmatrix} \sigma_{2,2} & \sigma_{2,4} & \sigma_{1,2} & \sigma_{2,3} \\ \sigma_{2,4} & \sigma_{4,4} & \sigma_{1,4} & \sigma_{3,4} \\ \sigma_{1,2} & \sigma_{1,4} & \sigma_{1,1} & \sigma_{1,3} \\ \sigma_{2,3} & \sigma_{3,4} & \sigma_{1,3} & \sigma_{3,3} \end{pmatrix}$$

This reverses the permutation:

```
Permute[%, {3, 1, 4, 2}] // MatrixForm
```

$$\begin{pmatrix} \sigma_{1,1} & \sigma_{1,2} & \sigma_{1,3} & \sigma_{1,4} \\ \sigma_{1,2} & \sigma_{2,2} & \sigma_{2,3} & \sigma_{2,4} \\ \sigma_{1,3} & \sigma_{2,3} & \sigma_{3,3} & \sigma_{3,4} \\ \sigma_{1,4} & \sigma_{2,4} & \sigma_{3,4} & \sigma_{4,4} \end{pmatrix}$$

```
(vTest = Table[σ_{Min[i,j],Max[i,j]}, {i, 3}, {j, 3}]) // MatrixForm
```

$$\begin{pmatrix} \sigma_{1,1} & \sigma_{1,2} & \sigma_{1,3} \\ \sigma_{1,2} & \sigma_{2,2} & \sigma_{2,3} \\ \sigma_{1,3} & \sigma_{2,3} & \sigma_{3,3} \end{pmatrix}$$

```
Permute[%, {3, 1, 2}] // MatrixForm
```

$$\begin{pmatrix} \sigma_{3,3} & \sigma_{1,3} & \sigma_{2,3} \\ \sigma_{1,3} & \sigma_{1,1} & \sigma_{1,2} \\ \sigma_{2,3} & \sigma_{1,2} & \sigma_{2,2} \end{pmatrix}$$

◻ *Matrix polynomial formatting routines*

```
PolyForm[m_, L_: L] :=
  Plus @@ Table[MatrixForm[Coefficient[m, L, i]] If[i =!= 0, L^i, ""],
    {i, -Max[Exponent[Flatten[m /. L → 1/L], L]], Max[Exponent[Flatten[m], L]]}];
```

◻ *Vector autocovariance generating function (from VMA representation);*

In the following, Θ is assumed to be a matrix polynomial in L; Ω is the covariance matrix. For more detail, see the last section of this notebook.

```
G[θ_, Ω_, z_] := (θ /. L → z).Ω.Transpose[θ /. L → z^-1];
```

Extended example with autocorrelated trades (Chapter 9)

Structural Model (Section 9.2)

```
qRule = qt_ :→ vt + β vt-1; mRule = mt_ :→ mt-1 + wt;
wRule = wt_ :→ λ vt + ut; pRule = pt_ :→ mt + c qt;
TableForm @ {qRule, mRule, wRule, pRule}
```

$q_{t\_} :\to v_t + \beta\, v_{t-1}$
$m_{t\_} :\to m_{t-1} + w_t$
$w_{t\_} :\to \lambda\, v_t + u_t$
$p_{t\_} :\to m_t + c\, q_t$

```
ΔpRule = Δpt_ :→ (pt /. pRule /. mRule /. wRule /. qRule) - (pt-1 /. pRule /. qRule);
Simplify[Δpt /. ΔpRule]
```

$u_t - c\,(\beta\, v_{-2+t} + v_{-1+t}) + \lambda\, v_t + c\,(\beta\, v_{-1+t} + v_t)$

```
nValues = {σv^2 → .8738, β → 0.381, c → 1, λ → 2}
```

$\{\sigma_v^2 \to 0.8738,\ \beta \to 0.381,\ c \to 1,\ \lambda \to 2\}$

VMA (not normalized)

The coefficients of current and lagged v_t in Δp_t may be tabulated as:

```
Coefficient[Δpt /. ΔpRule, vt-#] & /@ {0, 1, 2}
```

$\{c + \lambda,\ -c + c\,\beta,\ -c\,\beta\}$

This enables us to write: $\begin{pmatrix} \Delta p_t \\ q_t \end{pmatrix} = B_0 \begin{pmatrix} u_t \\ v_t \end{pmatrix} + B_1 \begin{pmatrix} u_{t-1} \\ v_{t-1} \end{pmatrix} + B_2 \begin{pmatrix} u_{t-2} \\ v_{t-2} \end{pmatrix}$ where

```
BRules =
  {B0 → {{1, λ + c}, {0, 1}}, B1 → {{0, c (β - 1)}, {0, β}}, B2 → {{0, -c β}, {0, 0}}};
{Map[MatrixForm, BRules, {2}]}
```

$\left\{\left\{B_0 \to \begin{pmatrix} 1 & c+\lambda \\ 0 & 1 \end{pmatrix},\ B_1 \to \begin{pmatrix} 0 & c\,(-1+\beta) \\ 0 & \beta \end{pmatrix},\ B_2 \to \begin{pmatrix} 0 & -c\,\beta \\ 0 & 0 \end{pmatrix}\right\}\right\}$

◻ *VMA (normalized so that the coefficient of the leading disturbance is the identity matrix)*

```
(B0i = Inverse[B0 /. BRules]) // MatrixForm
```

$$\begin{pmatrix} 1 & -c-\lambda \\ 0 & 1 \end{pmatrix}$$

Coefficients:

```
θRules = Table[θi → Simplify[(Bi /. BRules).B0i], {i, 0, 2}];
{Map[MatrixForm, θRules, 2]}
```

$$\left\{\left\{\theta_0 \to \begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}, \theta_1 \to \begin{pmatrix} 0 & c(-1+\beta) \\ 0 & \beta \end{pmatrix}, \theta_2 \to \begin{pmatrix} 0 & -c\beta \\ 0 & 0 \end{pmatrix}\right\}\right\}$$

NOTE: There is a typo in the first printing of the book. The correct expression for θ_1 is the one given above.

```
θ[L_] := θ0 + θ1 L + θ2 L^2;
```

Disturbances:

```
ϵRule = ϵt_ :→ Evaluate[B0.{{ut}, {vt}} /. BRules];
Map[MatrixForm, ϵRule, 1]
```

$$\epsilon_{t\_} :\to \begin{pmatrix} u_t + (c+\lambda)\, v_t \\ v_t \end{pmatrix}$$

Covariance matrix:

```
ΩRule = Ω → Simplify /@ B0.{{σu^2, 0}, {0, σv^2}}.Transpose[B0] /. BRules;
Ω /. ΩRule // MatrixForm
```

$$\begin{pmatrix} \sigma_u^2 + (c+\lambda)^2 \sigma_v^2 & (c+\lambda)\,\sigma_v^2 \\ (c+\lambda)\,\sigma_v^2 & \sigma_v^2 \end{pmatrix}$$

Random walk variance is in the upper left hand corner of

```
(θ[1].Ω.Transpose[θ[1]]) /. ΩRule /. θRules // Simplify // MatrixForm
```

$$\begin{pmatrix} \sigma_u^2 + \lambda^2 \sigma_v^2 & (1+\beta)\,\lambda\,\sigma_v^2 \\ (1+\beta)\,\lambda\,\sigma_v^2 & (1+\beta)^2 \sigma_v^2 \end{pmatrix}$$

◻ *VAR:*

Starting from the VMA $\begin{pmatrix} \Delta p_t \\ q_t \end{pmatrix} = \theta(L)\,\epsilon_t$, compute the VAR by inversion, i.e., $\theta(L)^{-1} \begin{pmatrix} \Delta p_t \\ q_t \end{pmatrix} = \epsilon_t$:

```
θInv = Inverse[θ[L] /. θRules] // Simplify; θInv // MatrixForm
```

$$\begin{pmatrix} 1 & \frac{c\,L\,(1+(-1+L)\,\beta)}{1+L\,\beta} \\ 0 & \frac{1}{1+L\,\beta} \end{pmatrix}$$

Series expansion to the fourth order:

```
s = Series[θInv, {L, 0, 4}]; s // MatrixForm
```

$$\begin{pmatrix} 1 & (c - c\,\beta)\,L + c\,\beta^2\,L^2 - c\,\beta^3\,L^3 + c\,\beta^4\,L^4 + O[L]^5 \\ 0 & 1 - \beta\,L + \beta^2\,L^2 - \beta^3\,L^3 + \beta^4\,L^4 + O[L]^5 \end{pmatrix}$$

Rules for VAR coefficient matrices:

```
φRules = Table[φi → (Coefficient[#, L, i] & /@ Normal[s]), {i, 0, 4}];
{Map[MatrixForm, φRules, {2}]}
```

$$\left\{\left\{\varphi_0 \to \begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}, \varphi_1 \to \begin{pmatrix} 0 & c - c\,\beta \\ 0 & -\beta \end{pmatrix}, \varphi_2 \to \begin{pmatrix} 0 & c\,\beta^2 \\ 0 & \beta^2 \end{pmatrix}, \varphi_3 \to \begin{pmatrix} 0 & -c\,\beta^3 \\ 0 & -\beta^3 \end{pmatrix}, \varphi_4 \to \begin{pmatrix} 0 & c\,\beta^4 \\ 0 & \beta^4 \end{pmatrix}\right\}\right\}$$

This VAR is of the form $\left(I + \varphi_1\,L + \varphi_2\,L^2 + ...\right)\begin{pmatrix} \Delta p_t \\ q_t \end{pmatrix} = \epsilon_t$. Alternatively,

$y_t = \phi_1\,y_{t-1} + \phi_2\,y_{t-2} + ... + \epsilon_t$ where $y_t \equiv \begin{pmatrix} \Delta p_t \\ q_t \end{pmatrix}$.

```
ϕRules = Table[ϕi → -φi /. φRules, {i, 1, 4}];
{Map[MatrixForm, ϕRules, {2}]}
```

$$\left\{\left\{\phi_1 \to \begin{pmatrix} 0 & -c + c\,\beta \\ 0 & \beta \end{pmatrix}, \phi_2 \to \begin{pmatrix} 0 & -c\,\beta^2 \\ 0 & -\beta^2 \end{pmatrix}, \phi_3 \to \begin{pmatrix} 0 & c\,\beta^3 \\ 0 & \beta^3 \end{pmatrix}, \phi_4 \to \begin{pmatrix} 0 & -c\,\beta^4 \\ 0 & -\beta^4 \end{pmatrix}\right\}\right\}$$

```
{Map[MatrixForm, ϕRules /. nValues, {2}]}
```

$$\left\{\left\{\phi_1 \to \begin{pmatrix} 0 & -0.619 \\ 0 & 0.381 \end{pmatrix}, \phi_2 \to \begin{pmatrix} 0 & -0.145161 \\ 0 & -0.145161 \end{pmatrix}, \phi_3 \to \begin{pmatrix} 0 & 0.0553063 \\ 0 & 0.0553063 \end{pmatrix}, \phi_4 \to \begin{pmatrix} 0 & -0.0210717 \\ 0 & -0.0210717 \end{pmatrix}\right\}\right\}$$

Impulse response functions

The VAR is of infinite order. We'll only be looking at the leading four terms:

$$\mathrm{VAR_{Rule}} = y_{t\_} :\to \sum_{k=1}^{4} \phi_k . y_{t-k} + \epsilon_t;\ y_t\ /.\ \mathrm{VAR_{Rule}}$$

$$\phi_1 . y_{-1+t} + \phi_2 . y_{-2+t} + \phi_3 . y_{-3+t} + \phi_4 . y_{-4+t} + \epsilon_t$$

(The "." indicates matrix multiplication.)

The following rules set $\epsilon_t = 0$ for $t \neq 0$, $y_t = 0$ for $t < 0$, and zero out matrix products involving zero.

```
CleanupRules = {y0 → ϵ0, yk_ :→ 0 /; k < 0, ϵk_ :→ 0 /; k ≠ 0, Dot[_, 0] :→ 0};
```

Subsequent to an initial disturbance of ϵ_0, the initial realization of y is:

```
y0 //. CleanupRules
```

ϵ_0

One period later ...

```
y1 /. VARRule //. CleanupRules
```

$\phi_1 . \epsilon_0$

Two periods later ...

```
y2 /. VARRule /. VARRule //. CleanupRules
```

$\phi_2 . \epsilon_0 + \phi_1 . \phi_1 . \epsilon_0$

The following rule obtains the impulse response function by automating the recursion

```
ψRule = ψt_ :→ (FixedPoint[# /. VARRule &, yt, t] //. CleanupRules);
```

```
ψ2 /. ψRule
```

$\phi_2 . \epsilon_0 + \phi_1 . \phi_1 . \epsilon_0$

```
ψ2 /. ψRule /. ϕRules /. ϵRule // Simplify
```

$\{\{-c\,\beta\,v_0\}, \{0\}\}$

The cumulative IR function is:

```
ΨRule = Ψt_ :→ Sum[ψi, {i, 0, t}];
```

$$\Psi_{Rule} = \Psi_{t\_} :\to \sum_{i=0}^{t} \psi_i;$$

If we substitute in the "correct" structural specification for ϵ_0, we get the "correct" long-run price change.

```
Ψ2 /. ΨRule /. ψRule /. ϕRules /. ϵRule // Simplify // MatrixForm
```

$$\begin{pmatrix} u_0 + \lambda v_0 \\ (1 + \beta)\, v_0 \end{pmatrix}$$

In practice, we can't identify u_0 and v_0, we we're left with:

```
ϵStatisticalRule = ϵt_ :> {{ϵΔp,t}, {ϵq,t}}; ϵ0 /. ϵStatisticalRule // MatrixForm
```

$$\begin{pmatrix} \epsilon_{\Delta p,0} \\ \epsilon_{q,0} \end{pmatrix}$$

And the cumulative impluse response function becomes:

```
Ψ2 /. ΨRule /. ψRule /. ϕRules /. ϵStatisticalRule // Simplify // MatrixForm
```

$$\begin{pmatrix} -C\,\epsilon_{q,0} + \epsilon_{\Delta p,0} \\ (1+\beta)\,\epsilon_{q,0} \end{pmatrix}$$

Contemporaneous effects and Cholesky decompositions (Section 9.4)

A note on notation and conventions. For a symmetric positive definite matrix V, the Cholesky decomposition is usually expressed as $V = L\,L'$ where L is lower triangular. In some references, though, and in *Mathematica*, the convention is $V = U'\,U$ where U is upper triangular. The book follows the latter (*Mathematica*) convention.

As an example, here's a 2×2 covariance matrix

```
V = {{σ1^2, σ12}, {σ12, σ2^2}};
V // MatrixForm
```

$$\begin{pmatrix} \sigma_1^2 & \sigma_{12} \\ \sigma_{12} & \sigma_2^2 \end{pmatrix}$$

```
F = Simplify[CholeskyDecomposition[V], {σ1^2 > 0, σ2^2 > 0, -Sqrt[σ1^2 σ2^2] < σ12 < Sqrt[σ1^2 σ2^2]}];
F // MatrixForm
```

$$\begin{pmatrix} \sqrt{\sigma_1^2} & \frac{\sigma_{12}}{\sqrt{\sigma_1^2}} \\ 0 & \sqrt{\sigma_2^2 - \frac{\sigma_{12}^2}{\sigma_1^2}} \end{pmatrix}$$

Verify that this recreates the original covariance matrix:

```
Transpose[F].F // Simplify // MatrixForm
```

$$\begin{pmatrix} \sigma_1^2 & \sigma_{12} \\ \sigma_{12} & \sigma_2^2 \end{pmatrix}$$

Here is the structural covariance matrix:

```
Map[MatrixForm, ΩRule, 1]
```

$$\Omega \to \begin{pmatrix} \sigma_u^2 + (c+\lambda)^2 \sigma_v^2 & (c+\lambda)\,\sigma_v^2 \\ (c+\lambda)\,\sigma_v^2 & \sigma_v^2 \end{pmatrix}$$

Its Cholesky decomposition with price changes "first" is:

```
F = Simplify[CholeskyDecomposition[Ω /. ΩRule], {c > 0, λ > 0, σu^2 > 0, σv^2 > 0}];
Transpose[F] // MatrixForm
```

$$\begin{pmatrix} \sqrt{\sigma_u^2 + (c+\lambda)^2 \sigma_v^2} & 0 \\ \frac{(c+\lambda)\,\sigma_v^2}{\sqrt{\sigma_u^2+(c+\lambda)^2 \sigma_v^2}} & \sqrt{\frac{\sigma_u^2\,\sigma_v^2}{\sigma_u^2+(c+\lambda)^2 \sigma_v^2}} \end{pmatrix}$$

This implies that a one unit shock to $\epsilon_{\Delta p,t}$ will have a contemporaneous effect on trades. This is inconsistent with the structural model. If we compute the Cholesky decomposition with "trades" first (by permutation) we have:

```
F = Permute[Simplify[CholeskyDecomposition[Permute[Ω /. ΩRule, {2, 1}]],
    {c > 0, λ > 0, σu^2 > 0, σv^2 > 0}], {2, 1}]; Transpose[F] // MatrixForm
```

$$\begin{pmatrix} \sqrt{\sigma_u^2} & (c+\lambda)\sqrt{\sigma_v^2} \\ 0 & \sqrt{\sigma_v^2} \end{pmatrix}$$

```
Transpose[F].{{0}, {1}} // MatrixForm
```

$$\begin{pmatrix} (c+\lambda)\sqrt{\sigma_v^2} \\ \sqrt{\sigma_v^2} \end{pmatrix}$$

Random-walk variance

Starting from the VMA $\begin{pmatrix} \Delta p_t \\ q_t \end{pmatrix} = \theta(L)\,\epsilon_t$, compute the VAR by inversion, i.e., $\theta(L)^{-1} \begin{pmatrix} \Delta p_t \\ q_t \end{pmatrix} = \epsilon_t$:

$$\begin{pmatrix} \sigma_w^2 & \cdots \\ \vdots & \square \end{pmatrix} = \theta(1)\,\Omega\,\theta(1)'$$

θ is identified (or get $\theta(1) = \varphi(1)^{-1}$).

```
θ[1] /. θRules // Simplify // MatrixForm
```

$$\begin{pmatrix} 1 & -c \\ 0 & 1+\beta \end{pmatrix}$$

```
{{1, -c}}.(Ω /. ΩRule).Transpose[{{1, -c}}] // Simplify // MatrixForm
```

$$\left(\sigma_u^2 + \lambda^2 \sigma_v^2 \right)$$

In practice, we'd be working from an estimated covariance matrix:

```
{{1, -c}}.V.Transpose[{{1, -c}}] // Simplify // MatrixForm
```

$$\left(\sigma_1^2 + c^2 \sigma_2^2 - 2\, c\, \sigma_{12} \right)$$

Resolve contemporaneous effects with Cholesky decompositions.

"Prices first"

```
F = Simplify[CholeskyDecomposition[Ω /. ΩRule], {c > 0, λ > 0, σu^2 > 0, σv^2 > 0}];
Transpose[F] // MatrixForm
```

$$\begin{pmatrix} \sqrt{\sigma_u^2 + (c+\lambda)^2 \sigma_v^2} & 0 \\ \frac{(c+\lambda)\, \sigma_v^2}{\sqrt{\sigma_u^2 + (c+\lambda)^2 \sigma_v^2}} & \sqrt{\frac{\sigma_u^2 \sigma_v^2}{\sigma_u^2 + (c+\lambda)^2 \sigma_v^2}} \end{pmatrix}$$

```
d = First @ {{1, -c}}.Transpose[F] // Simplify
```

$$\left\{ \frac{\sigma_u^2 + \lambda\, (c+\lambda)\, \sigma_v^2}{\sqrt{\sigma_u^2 + (c+\lambda)^2 \sigma_v^2}},\ -c \sqrt{\frac{\sigma_u^2 \sigma_v^2}{\sigma_u^2 + (c+\lambda)^2 \sigma_v^2}} \right\}$$

```
Plus @@ (d^2) // Simplify
```

$$\sigma_u^2 + \lambda^2 \sigma_v^2$$

Trades "first"

```
F = Permute[Simplify[CholeskyDecomposition[Permute[Ω /. ΩRule, {2, 1}]],
    {c > 0, λ > 0, σu^2 > 0, σv^2 > 0}], {2, 1}]; Transpose[F] // MatrixForm
```

$$\begin{pmatrix} \sqrt{\sigma_u^2} & (c+\lambda) \sqrt{\sigma_v^2} \\ 0 & \sqrt{\sigma_v^2} \end{pmatrix}$$

```
d = First @ {{1, -c}}.Transpose[F] // Simplify
```

$$\left\{ \sqrt{\sigma_u^2},\ \lambda \sqrt{\sigma_v^2} \right\}$$

```
Plus @@ (d^2) // Simplify
```

$\sigma_u^2 + \lambda^2 \sigma_v^2$

Pricing error

Structural model:

```
p_t - m_t /. p_Rule
```

$c\, q_t$

$\Rightarrow \mathrm{Var}(s_t) = c^2\, \mathrm{Var}(q_t) = c^2 (1+\beta)^2 \sigma_v^2$

Lower bound

```
b = (θ_1 + θ_2) + θ_2
```

$\theta_1 + 2\,\theta_2$

```
b /. θ_Rules // Simplify // MatrixForm
```

$$\begin{pmatrix} 0 & -c\,(1+\beta) \\ 0 & \beta \end{pmatrix}$$

```
(sv = b.Ω.Transpose[b] /. Ω_Rule /. θ_Rules // Simplify) // MatrixForm
```

$$\begin{pmatrix} (c + c\,\beta)^2 \sigma_v^2 & -c\,\beta\,(1+\beta)\,\sigma_v^2 \\ -c\,\beta\,(1+\beta)\,\sigma_v^2 & \beta^2 \sigma_v^2 \end{pmatrix}$$

```
sv /. nValues // MatrixForm
```

$$\begin{pmatrix} 1.66648 & -0.459759 \\ -0.459759 & 0.126842 \end{pmatrix}$$

Problems

Exercise 9.1 (Glosten and Harris)

Definitions:

```
m_Rule = m_t_ :> m_{t-1} + w_t;
w_Rule = w_t_ :> λ_0 q_t + λ_1 Q_t + u_t;
p_Rule = p_t_ :> m_t + c_1 q_t + c_2 Q_t;
```

With these definitions, m_t, w_t and p_t are (respectively):

```
TableForm[{ mt /. mRule, wt /. wRule, pt /. pRule}]
```

$m_{-1+t} + w_t$
$u_t + q_t \lambda_0 + Q_t \lambda_1$
$m_t + c_1 q_t + c_2 Q_t$

The price change at time t is $\Delta p_t =$

```
ΔpRule = Δpt_ :→ (pt /. pRule /. mRule /. wRule) − (pt-1 /. pRule);
Simplify[Δpt /. ΔpRule]
```

$c_1 (-q_{-1+t} + q_t) + c_2 (-Q_{-1+t} + Q_t) + u_t + q_t \lambda_0 + Q_t \lambda_1$

The vector of variables is:

```
yRule = yt_ :→ Transpose[{{Δpt, qt, Qt}}]; MatrixForm[yt /. yRule]
```

$$\begin{pmatrix} \Delta p_t \\ q_t \\ Q_t \end{pmatrix}$$

and the disturbances are $\epsilon_t =$

```
ϵRule = ϵt_ :→ Transpose[{{ut, qt, Qt}}]; MatrixForm[ϵt /. ϵRule]
```

$$\begin{pmatrix} u_t \\ q_t \\ Q_t \end{pmatrix}$$

With substitutions, the vector of system variables is:

```
(yt /. yRule /. ΔpRule) // MatrixForm // Simplify
```

$$\begin{pmatrix} c_1 (-q_{-1+t} + q_t) + c_2 (-Q_{-1+t} + Q_t) + u_t + q_t \lambda_0 + Q_t \lambda_1 \\ q_t \\ Q_t \end{pmatrix}$$

The MA model is $y_t = \theta_0 \epsilon_t + \theta_1 \epsilon_{t-1}$ where the coefficient matrices are:

```
ClearAll[θ];
θRules = Table[Rule[θs, Table[Coefficient[(yt /. yRule /. ΔpRule)[[i, 1]],
      (ϵt-s /. ϵRule)[[j, 1]]], {i, 3}, {j, 3}]], {s, 0, 1}];
{Map[MatrixForm, θRules, {2}]}
```

$$\left\{\left\{\theta_0 \to \begin{pmatrix} 1 & c_1 + \lambda_0 & c_2 + \lambda_1 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}, \theta_1 \to \begin{pmatrix} 0 & -c_1 & -c_2 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}\right\}\right\}$$

To build the covariance matrix, we need to work out $\mathrm{Cov}(q_t, Q_t) = E[Q_t \,\mathrm{Sign}(Q_t)] = E\,|\,Q_t\,|$

```
ClearAll[Ω]
```

If $Q_t \sim N(0, \sigma_Q^2)$, then $E[\,|Q|\,] =$

```
Simplify[∫_{-∞}^{∞} Abs[x] PDF[NormalDistribution[0, σ_Q], x] ⅆx, {σ_Q ∈ Reals, Re[(σ_Q)^2] > 0}]
```

$$\sqrt{\frac{2}{\pi}}\ \sigma_Q$$

So the disturbance covariance matrix is:

```
Ω_Rule = Ω → ({{σ_u^2, 0, 0}, {0, 1, σ_{q,Q}}, {0, σ_{q,Q}, σ_Q^2}} /. σ_{q,Q} → Sqrt[2 σ_Q^2 / π]);
Map[MatrixForm, Ω_Rule, 1]
```

$$\Omega \to \begin{pmatrix} \sigma_u^2 & 0 & 0 \\ 0 & 1 & \sqrt{\frac{2}{\pi}}\ \sqrt{\sigma_Q^2} \\ 0 & \sqrt{\frac{2}{\pi}}\ \sqrt{\sigma_Q^2} & \sigma_Q^2 \end{pmatrix}$$

The random-walk variance is $\sigma_w^2 =$

```
Simplify[(θ_0 + θ_1).Ω.Transpose[θ_0 + θ_1] /. θ_Rules /. Ω_Rule, {σ_Q > 0, σ_ ∈ Reals}][[1, 1]]
```

$$\sigma_u^2 + \lambda_0^2 + 2\sqrt{\frac{2}{\pi}}\ \sqrt{\sigma_Q^2}\ \lambda_0\,\lambda_1 + \sigma_Q^2\,\lambda_1^2$$

- *Decomposition with q_t first:*

```
F1 = Simplify[Permute[CholeskyDecomposition[Permute[Ω /. Ω_Rule, {2, 3, 1}]],
    {3, 1, 2}], {σ_Q^2 > 0, σ_u^2 > 0}];
F1 // MatrixForm
```

$$\begin{pmatrix} \sqrt{\sigma_u^2} & 0 & 0 \\ 0 & 1 & \sqrt{\frac{2}{\pi}}\ \sqrt{\sigma_Q^2} \\ 0 & 0 & \sqrt{\frac{-2+\pi}{\pi}}\ \sqrt{\sigma_Q^2} \end{pmatrix}$$

```
VarDecomp1 = ((θ_0 + θ_1).Transpose[F1] /. θ_Rules // Simplify)[[1]]
```

$$\left\{\sqrt{\sigma_u^2},\ \lambda_0 + \sqrt{\frac{2}{\pi}}\ \sqrt{\sigma_Q^2}\ \lambda_1,\ \sqrt{\frac{-2+\pi}{\pi}}\ \sqrt{\sigma_Q^2}\ \lambda_1\right\}$$

The variance components corresponding to u_t, q_t and Q_t are:

```
{VarDecomp1^2 // Simplify} // TableForm
```

$$\sigma_u^2 \qquad \left(\lambda_0 + \sqrt{\frac{2}{\pi}}\sqrt{\sigma_Q^2}\,\lambda_1\right)^2 \qquad \frac{(-2+\pi)\,\sigma_Q^2\,\lambda_1^2}{\pi}$$

Verify that they add up to the correct σ_w^2:

```
Plus @@ VarDecomp1^2 // Simplify
```

$$\sigma_u^2 + \lambda_0^2 + 2\sqrt{\frac{2}{\pi}}\sqrt{\sigma_Q^2}\,\lambda_0\,\lambda_1 + \sigma_Q^2\,\lambda_1^2$$

□ *Decomposition with Q_t first:*

```
F2 = Simplify[Permute[CholeskyDecomposition[Permute[Ω /. ΩRule, {3, 2, 1}]],
    {3, 2, 1}], {σQ > 0, σu > 0}];
F2 // MatrixForm
```

$$\begin{pmatrix} \sqrt{\sigma_u^2} & 0 & 0 \\ 0 & \sqrt{\frac{-2+\pi}{\pi}} & 0 \\ 0 & \sqrt{\frac{2}{\pi}} & \sqrt{\sigma_Q^2} \end{pmatrix}$$

```
VarDecomp2 = ((θ0 + θ1).Transpose[F2] /. θRules // Simplify)[[1]]
```

$$\left\{\sqrt{\sigma_u^2},\ \sqrt{\frac{-2+\pi}{\pi}}\,\lambda_0,\ \sqrt{\frac{2}{\pi}}\,\lambda_0 + \sqrt{\sigma_Q^2}\,\lambda_1\right\}$$

```
{VarDecomp2^2} // Simplify // TableForm
```

$$\sigma_u^2 \qquad \frac{(-2+\pi)\,\lambda_0^2}{\pi} \qquad \left(\sqrt{\frac{2}{\pi}}\,\lambda_0 + \sqrt{\sigma_Q^2}\,\lambda_1\right)^2$$

... and verify:

```
Plus @@ VarDecomp2^2 // Simplify
```

$$\sigma_u^2 + \lambda_0^2 + 2\sqrt{\frac{2}{\pi}}\sqrt{\sigma_Q^2}\,\lambda_0\,\lambda_1 + \sigma_Q^2\,\lambda_1^2$$

Exercise 9.2 (Madhavan, Richardson and Roomans)

The model is:

```
qRule = q[t_] :> v[t] + β q[t-1];
mRule = m[t_] :> m[t-1] + w[t];
wRule = w[t_] :> λ v[t] + u[t];
pRule = p[t_] :> m[t] + c q[t];
```

With these rules, q_t, m_t, w_t, and p_t are:

```
{q[t] /. qRule, m[t] /. mRule, w[t] /. wRule, p[t] /. pRule} // TableForm
```

$\beta\, q_{-1+t} + v_t$
$m_{-1+t} + w_t$
$u_t + \lambda\, v_t$
$m_t + c\, q_t$

The price change is Δp_t=

```
ΔpRule = Δp[t_] :> Δp[t] → (p[t] /. pRule /. mRule /. wRule) − (p[t-1] /. pRule);
Simplify[Δp[t] /. ΔpRule]
```

$\Delta p_t \to -c\, q_{-1+t} + c\, q_t + u_t + \lambda\, v_t$

The system variables are y_t =

```
yRule = y[t_] :> Transpose[{{Δp[t], q[t]}}]; y[t] /. yRule // MatrixForm
```

$$\begin{pmatrix} \Delta p_t \\ q_t \end{pmatrix}$$

The vector of disturbances is:

```
ϵRule = ϵ[t_] :> Transpose[{{u[t], v[t]}}]; ϵ[t] /. ϵRule // MatrixForm
```

$$\begin{pmatrix} u_t \\ v_t \end{pmatrix}$$

```
ClearAll[θ]
```

With substitutions, the vector of system variables becomes y_t =

```
{{Δp[t]}, {q[t]}} /. Δp[t] → (p[t] /. pRule /. mRule /. wRule) − (p[t-1] /. pRule) /. q[t] → (q[t] /. qRule) //
  Simplify // MatrixForm
```

$$\begin{pmatrix} c\,(-1+\beta)\, q_{-1+t} + u_t + (c+\lambda)\, v_t \\ \beta\, q_{-1+t} + v_t \end{pmatrix}$$

This is a first-order vector autoregressive process: $y_t = \phi\, y_{t-1} + \Theta \epsilon_t$ where

```
ϕRule = ϕ → {{0, c (-1 + β)}, {0, β}}; MatrixForm /@ ϕRule
```

$$\phi \to \begin{pmatrix} 0 & c\,(-1+\beta) \\ 0 & \beta \end{pmatrix}$$

and

```
ΘRule = Θ → {{1, (c + λ)}, {0, 1}}; MatrixForm /@ ΘRule
```

$$\Theta \to \begin{pmatrix} 1 & c+\lambda \\ 0 & 1 \end{pmatrix}$$

It may be put in vector moving average (VMA) form as: $y_t = \underbrace{(I - \phi L)^{-1}\,\Theta}_{\text{VMA coefficients}} \epsilon_t$. We could obtain the VMA coefficient matrices by doing the series expansion. Here, though, to compute the random-walk variance, we just need the sum of the moving average coefficients, and $(I - \phi)^{-1}\,\Theta =$

```
maSum = Inverse[IdentityMatrix[2] - (ϕ /. ϕRule)].(Θ /. ΘRule) // Simplify;
maSum // MatrixForm
```

$$\begin{pmatrix} 1 & \lambda \\ 0 & \frac{1}{1-\beta} \end{pmatrix}$$

To compute the random-walk variance σ_w^2, take the upper left hand entry of $\left[(I - \phi)^{-1}\,\Theta\right] \Omega \left[(I - \phi)^{-1}\,\Theta\right]'$

```
maSum.{{σu², 0}, {0, σv²}}.Transpose[maSum] // Simplify // MatrixForm
```

$$\begin{pmatrix} \sigma_u^2 + \lambda^2\,\sigma_v^2 & \frac{\lambda\,\sigma_v^2}{1-\beta} \\ \frac{\lambda\,\sigma_v^2}{1-\beta} & \frac{\sigma_v^2}{(-1+\beta)^2} \end{pmatrix}$$

The quantity $\sigma_u^2 + \lambda^2\,\sigma_v^2$ summarizes the public information and trade-related components of the random-walk.

Multiple Securities (Chapter 10)

Two securities in a Roll model, not cointegrated, prices only. (Chapter 10, Section 10.1)

Model

```
pRule = p_{i_,t_} :→ m_{i,t} + c q_{i,t};
mRule = m_{i_,t_} :→ m_{i,t-1} + w_{i,t};
wRule = w_{i_,t_} :→ u_{i,t};
ΔpRule = Δp_{i_,t_} :→ (p_{i,t} /. pRule /. mRule /. wRule) - (p_{i,t-1} /. pRule);
Δp_{t_} := {{Δp_{1,t}}, {Δp_{2,t}}};
```

```
Δp_t /. ΔpRule
```

$\{\{-c\, q_{1,-1+t} + c\, q_{1,t} + u_{1,t}\},\ \{-c\, q_{2,-1+t} + c\, q_{2,t} + u_{2,t}\}\}$

Compute autocovariances

We will obtain $\Gamma_k = E\Delta p_t\ \Delta p_{t-k}$, where the following rules are used to eliminate terms in the expectation with zero cross-products. (Note: these are general rules; some of them are left over from the univariate case.)

```
ℰRules = {ℰ[q_^2] → 1, ℰ[u_^2] → σu^2, ℰ[q_ u_] → 0, ℰ[q_{t_} q_{s_}] :→ 0 /; t =!= s,
   ℰ[u_{t_} u_{s_}] :→ 0 /; t =!= s, ℰ[q_{_,_} u_{_,_}] → 0, ℰ[u_{_,_}^2] :→ σu^2, ℰ[u_{_,t_} u_{_,s_}] → 0,
   ℰ[q_{_,t_} q_{_,s_}] :→ ρ /; t === s, ℰ[q_{_,t_} q_{_,s_}] :→ 0 /; t =!= s, ℰ[q_{_,_}^2] → 1,
ℰ[a_ + b_] :→ ℰ[a] + ℰ[b],
ℰ[c_] :→ c /; NumberQ[c],
   ℰ[c_Symbol] :→ c /; MemberQ[Attributes[c], Constant],
   ℰ[c_ a_] :→ c ℰ[a] /; NumberQ[c],
ℰ[c_Symbol a_] :→ c ℰ[a] /; MemberQ[Attributes[c], Constant],
ℰ[c_Symbol^n_Integer a_] :→ c^n ℰ[a] /; MemberQ[Attributes[c], Constant]};
```

```
ℰRules // TableForm
```

$$
\begin{array}{l}
\mathcal{E}[q_\_^2] \to 1 \\
\mathcal{E}[u_\_^2] \to \sigma_u^2 \\
\mathcal{E}[q\_\ u\_] \to 0 \\
\mathcal{E}[q_{s\_}\ q_{t\_}] :\to 0 \ /;\ t =!= s \\
\mathcal{E}[u_{s\_}\ u_{t\_}] :\to 0 \ /;\ t =!= s \\
\mathcal{E}[q_{\_,\_}\ u_{\_,\_}] \to 0 \\
\mathcal{E}[u_{\_,\_}^2] :\to \sigma_u^2 \\
\mathcal{E}[u_{\_,s\_}\ u_{\_,t\_}] \to 0 \\
\mathcal{E}[q_{\_,s\_}\ q_{\_,t\_}] :\to \rho \ /;\ t === s \\
\mathcal{E}[q_{\_,s\_}\ q_{\_,t\_}] :\to 0 \ /;\ t =!= s \\
\mathcal{E}[q_{\_,\_}^2] \to 1 \\
\mathcal{E}[a\_ + b\_] :\to \mathcal{E}[a] + \mathcal{E}[b] \\
\mathcal{E}[c\_] :\to c \ /;\ \text{NumberQ}[c] \\
\mathcal{E}[c\_\text{Symbol}] :\to c \ /;\ \text{MemberQ}[\text{Attributes}[c], \text{Constant}] \\
\mathcal{E}[a\_\ c\_] :\to c\ \mathcal{E}[a] \ /;\ \text{NumberQ}[c] \\
\mathcal{E}[a\_\ c\_\text{Symbol}] :\to c\ \mathcal{E}[a] \ /;\ \text{MemberQ}[\text{Attributes}[c], \text{Constant}] \\
\mathcal{E}[a\_\ c\_\text{Symbol}^{n\_\text{Integer}}] :\to c^n\ \mathcal{E}[a] \ /;\ \text{MemberQ}[\text{Attributes}[c], \text{Constant}]
\end{array}
$$

```
SetAttributes[ℰ, Listable]
```

$$\Gamma_{\text{Rule}} = \Gamma_{k\_} :\to \left(\mathcal{E}\left[\text{Expand}\left[\left(\Delta p_t \ /.\ \Delta p_{\text{Rule}}\right).\text{Transpose}\left[\Delta p_{t-k} \ /.\ \Delta p_{\text{Rule}}\right]\right]\right] \ //.\ \mathcal{E}_{\text{Rules}}\right);$$

$$\Gamma_0 \ /.\ \Gamma_{\text{Rule}} \ //\ \text{MatrixForm}$$

$$\begin{pmatrix} 2\,c^2 + \sigma_u^2 & 2\,c^2\,\rho \\ 2\,c^2\,\rho & 2\,c^2 + \sigma_u^2 \end{pmatrix}$$

Γ_0, Γ_1 and Γ_2 are:

$$\text{Map}[\text{MatrixForm}, \text{Table}[\Gamma_k \ /.\ \Gamma_{\text{Rule}}, \{k, 0, 2\}], 1]$$

$$\left\{\begin{pmatrix} 2\,c^2 + \sigma_u^2 & 2\,c^2\,\rho \\ 2\,c^2\,\rho & 2\,c^2 + \sigma_u^2 \end{pmatrix}, \begin{pmatrix} -c^2 & -c^2\,\rho \\ -c^2\,\rho & -c^2 \end{pmatrix}, \begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}\right\}$$

Note that $\Gamma_2 = 0$. We can verify that $\Gamma_k = 0$ for $|k| \geq 2$.

Here are some trial numerical values:

$$\text{nValues} = \{\sigma_u^2 \to 1, c \to 10, \rho \to .9\}$$

$$\{\sigma_u^2 \to 1, c \to 10, \rho \to 0.9\}$$

$$\text{Map}[\text{MatrixForm}, \text{Table}[\Gamma_k \ /.\ \Gamma_{\text{Rule}}, \{k, 0, 1\}], 1] \ /.\ \text{nValues}$$

$$\left\{\begin{pmatrix} 201 & 180. \\ 180. & 201 \end{pmatrix}, \begin{pmatrix} -100 & -90. \\ -90. & -100 \end{pmatrix}\right\}$$

Set up VMA form

Since the autocovariances vanish after lag 1, we have a VMA(1) structure:

```
θ[L_] := IdentityMatrix[2] + Table[θ_{i,j}, {i, 2}, {j, 2}] L;
θ[L] // PolyForm
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix} + L \begin{pmatrix} \theta_{1,1} & \theta_{1,2} \\ \theta_{2,1} & \theta_{2,2} \end{pmatrix}$$

```
Ω = {{σ_1^2, σ_{1,2}}, {σ_{1,2}, σ_2^2}}; Ω // MatrixForm
```

$$\begin{pmatrix} \sigma_1^2 & \sigma_{1,2} \\ \sigma_{1,2} & \sigma_2^2 \end{pmatrix}$$

Autocovariance generating function:

```
g[z_] := θ[z].Ω.Transpose[θ[z^-1]]
```

We'll obtain the Γ_k for the VMA representation by picking off the appropriate coefficient in the AGF.

```
ΓVMA_Rule = Γ_k_ :→ Simplify[Coefficient[g[z], z, k]];
Map[MatrixForm, Table[Γ_k /. ΓVMA_Rule, {k, 0, 1}]] // MatrixForm
```

$$\begin{pmatrix} \begin{pmatrix} \sigma_1^2 + \sigma_1^2 \theta_{1,1}^2 + \sigma_2^2 \theta_{1,2}^2 + 2\,\theta_{1,1}\theta_{1,2}\sigma_{1,2} & \sigma_{1,2} + \theta_{1,2}\left(\sigma_2^2 \theta_{2,2} + \theta_{2,1}\sigma_{1,2}\right) + \theta_{1,1} \\ \sigma_{1,2} + \theta_{1,2}\left(\sigma_2^2\theta_{2,2} + \theta_{2,1}\sigma_{1,2}\right) + \theta_{1,1}\left(\sigma_1^2\theta_{2,1} + \theta_{2,2}\sigma_{1,2}\right) & \sigma_2^2 + \sigma_1^2\theta_{2,1}^2 + \sigma_2^2\theta_{2,2}^2 + 2\,\theta_{2,1}\theta_{2,2}\sigma_{1,2} \end{pmatrix} \\ \begin{pmatrix} \sigma_1^2\theta_{1,1} + \theta_{1,2}\sigma_{1,2} & \sigma_2^2\theta_{1,2} + \theta_{1,1}\sigma_{1,2} \\ \sigma_1^2\theta_{2,1} + \theta_{2,2}\sigma_{1,2} & \sigma_2^2\theta_{2,2} + \theta_{2,1}\sigma_{1,2} \end{pmatrix} \end{pmatrix}$$

Solve for VMA parameters by equating the two representations:

```
eq = Flatten @ MapThread[Equal, {{Γ_0, Γ_1} /. Γ_Rule, {Γ_0, Γ_1} /. ΓVMA_Rule}, 3];
eq // TableForm
```

$$2\,c^2 + \sigma_u^2 = \sigma_1^2 + \sigma_1^2\theta_{1,1}^2 + \sigma_2^2\theta_{1,2}^2 + 2\,\theta_{1,1}\theta_{1,2}\sigma_{1,2}$$
$$2\,c^2 \rho = \sigma_{1,2} + \theta_{1,2}\left(\sigma_2^2\theta_{2,2} + \theta_{2,1}\sigma_{1,2}\right) + \theta_{1,1}\left(\sigma_1^2\theta_{2,1} + \theta_{2,2}\sigma_{1,2}\right)$$
$$2\,c^2 \rho = \sigma_{1,2} + \theta_{1,2}\left(\sigma_2^2\theta_{2,2} + \theta_{2,1}\sigma_{1,2}\right) + \theta_{1,1}\left(\sigma_1^2\theta_{2,1} + \theta_{2,2}\sigma_{1,2}\right)$$
$$2\,c^2 + \sigma_u^2 = \sigma_2^2 + \sigma_1^2\theta_{2,1}^2 + \sigma_2^2\theta_{2,2}^2 + 2\,\theta_{2,1}\theta_{2,2}\sigma_{1,2}$$
$$-c^2 = \sigma_1^2\theta_{1,1} + \theta_{1,2}\sigma_{1,2}$$
$$-c^2\rho = \sigma_2^2\theta_{1,2} + \theta_{1,1}\sigma_{1,2}$$
$$-c^2\rho = \sigma_1^2\theta_{2,1} + \theta_{2,2}\sigma_{1,2}$$
$$-c^2 = \sigma_2^2\theta_{2,2} + \theta_{2,1}\sigma_{1,2}$$

The third equation is redundant, so ...

```
e2 = Drop[eq, {3}]
```

$$\{2c^2+\sigma_u^2==\sigma_1^2+\sigma_1^2\theta_{1,1}^2+\sigma_2^2\theta_{1,2}^2+2\theta_{1,1}\theta_{1,2}\sigma_{1,2},$$
$$2c^2\rho==\sigma_{1,2}+\theta_{1,2}\left(\sigma_2^2\theta_{2,2}+\theta_{2,1}\sigma_{1,2}\right)+\theta_{1,1}\left(\sigma_1^2\theta_{2,1}+\theta_{2,2}\sigma_{1,2}\right),$$
$$2c^2+\sigma_u^2==\sigma_2^2+\sigma_1^2\theta_{2,1}^2+\sigma_2^2\theta_{2,2}^2+2\theta_{2,1}\theta_{2,2}\sigma_{1,2},\ -c^2==\sigma_1^2\theta_{1,1}+\theta_{1,2}\sigma_{1,2},$$
$$-c^2\rho==\sigma_2^2\theta_{1,2}+\theta_{1,1}\sigma_{1,2},\ -c^2\rho==\sigma_1^2\theta_{2,1}+\theta_{2,2}\sigma_{1,2},\ -c^2==\sigma_2^2\theta_{2,2}+\theta_{2,1}\sigma_{1,2}\}$$

```
parms = {θ1,1, θ1,2, θ2,1, θ2,2, σ1^2, σ2^2, σ1,2};
```

We'll attempt only a numerical solution here:

```
s = NSolve[e2 /. nValues, parms]; s // TableForm
```

$\theta_{1,1}\to-1.1501$	$\theta_{1,2}\to0.22006$	$\theta_{2,1}\to0.22006$	$\theta_{2,2}\to-1.1501$	$\sigma_1^2\to105.796$	$\sigma_2^2\to1$
$\theta_{1,1}\to-0.82994$	$\theta_{1,2}\to-0.100096$	$\theta_{2,1}\to-0.100096$	$\theta_{2,2}\to-0.82994$	$\sigma_1^2\to108.997$	$\sigma_2^2\to1$
$\theta_{1,1}\to-1.22269$	$\theta_{1,2}\to0.147465$	$\theta_{2,1}\to0.147465$	$\theta_{2,2}\to-1.22269$	$\sigma_1^2\to92.0027$	$\sigma_2^2\to9$
$\theta_{1,1}\to-0.902535$	$\theta_{1,2}\to-0.172692$	$\theta_{2,1}\to-0.172692$	$\theta_{2,2}\to-0.902535$	$\sigma_1^2\to95.2042$	$\sigma_2^2\to9$

Consider the matrix

```
DD = (θ[L] /. L → 1 / L) * L // Simplify;
DD // MatrixForm
```

$$\begin{pmatrix} L+\theta_{1,1} & \theta_{1,2} \\ \theta_{2,1} & L+\theta_{2,2} \end{pmatrix}$$

For the VMA to be invertible, the roots of the determinental equation Det[DD]=0 must lie inside the unit circle. We check:

```
Table[Solve[Det[DD] == 0 /. s[[i]], L] // Flatten, {i, 4}] // MatrixForm
```

$$\begin{pmatrix} L\to0.930036 & L\to1.37016 \\ L\to0.729844 & L\to0.930036 \\ L\to1.07523 & L\to1.37016 \\ L\to0.729844 & L\to1.07523 \end{pmatrix}$$

Only the second solution is invertible:

```
sInvertible = s[[2]]
```

$$\{\theta_{1,1}\to-0.82994,\ \theta_{1,2}\to-0.100096,\ \theta_{2,1}\to-0.100096,$$
$$\theta_{2,2}\to-0.82994,\ \sigma_1^2\to108.997,\ \sigma_2^2\to108.997,\ \sigma_{1,2}\to95.2958\}$$

```
Table[θi,j, {i, 2}, {j, 2}] /. sInvertible // MatrixForm
```

$$\begin{pmatrix} -0.82994 & -0.100096 \\ -0.100096 & -0.82994 \end{pmatrix}$$

```
Ω /. sInvertible // MatrixForm
```

$$\begin{pmatrix} 108.997 & 95.2958 \\ 95.2958 & 108.997 \end{pmatrix}$$

Verify the computation of the random-walk variance:

```
θ[1].Ω.Transpose[θ[1]] /. sInvertible // Chop // MatrixForm
```

$$\begin{pmatrix} 1. & 0 \\ 0 & 1. \end{pmatrix}$$

Variance decomposition.

```
(F = CholeskyDecomposition[Ω /. sInvertible]) // MatrixForm
```

$$\begin{pmatrix} 10.4402 & 9.12779 \\ 0. & 5.06762 \end{pmatrix}$$

```
NumberForm[Transpose[F] // MatrixForm, {5, 2}]
```

$$\begin{pmatrix} 10.44 & 0.00 \\ 9.13 & 5.07 \end{pmatrix}$$

```
NumberForm[θ[1].Transpose[F] /. sInvertible // MatrixForm, {5, 3}]
```

$$\begin{pmatrix} 0.862 & -0.507 \\ 0.507 & 0.862 \end{pmatrix}$$

```
r = (θ[1].Transpose[F] /. sInvertible)[[1]]
```

{0.861799, -0.50725}

r^2

{0.742698, 0.257302}

Cointegrated model (Chapter 10, section 10.2)

Model

$$p_{Rule} = \{p_{1,t\_} \mapsto m_t + c\, q_t,\ p_{2,t\_} \mapsto m_{t-1}\};$$
$$m_{Rule} = m_{t\_} \mapsto m_{t-1} + u_t;$$
$$\Delta p_{Rule} = \Delta p_{i\_,t\_} \mapsto (p_{i,t} \text{ /. } p_{Rule} \text{ /. } m_{Rule}) - (p_{i,t-1} \text{ /. } p_{Rule});$$
$$\Delta p_{t\_} := \{\{\Delta p_{1,t}\},\ \{\Delta p_{2,t}\}\};$$

The vector of price differences is:

```
Δp_t /. Δp_Rule // MatrixForm
```

$$\begin{pmatrix} -c\, q_{-1+t} + c\, q_t + u_t \\ u_{-1+t} \end{pmatrix}$$

The difference between the two prices is:

```
((p_{1,t} /. p_Rule /. m_Rule) - (p_{2,t} /. p_Rule))
```

$c\, q_t + u_t$

◻ *Structural VMA representation: VMA I*

```
θ[L_] := θ_0 + θ_1 L
```

```
ϵ_Rule = ϵ_t_ :→ {{u_t}, {q_t}};
```

```
θ_Rules =
  Table[Rule[θ_s, Table[Coefficient[(Δp_t /. Δp_Rule)[[i, 1]], (ϵ_{t-s} /. ϵ_Rule)[[j, 1]]],
     {i, 2}, {j, 2}]], {s, 0, 1}];
{Map[MatrixForm, θ_Rules, {2}]} // TableForm
```

$\theta_0 \to \begin{matrix} 1 & c \\ 0 & 0 \end{matrix} \quad \theta_1 \to \begin{matrix} 0 & -c \\ 1 & 0 \end{matrix}$

```
Clear[Ω];
Ω_Rule = Ω → {{σ_u^2, 0}, {0, 1}}
```

$\Omega \to \{\{\sigma_u^2, 0\}, \{0, 1\}\}$

Autocovariance generating function:

```
agfI = G[θ[L] /. θ_Rules, Ω /. Ω_Rule, z];
PolyForm[agfI, z]
```

$$z \begin{pmatrix} -c^2 & 0 \\ \sigma_u^2 & 0 \end{pmatrix} + \frac{\begin{pmatrix} -c^2 & \sigma_u^2 \\ 0 & 0 \end{pmatrix}}{z} + \begin{pmatrix} 2c^2 + \sigma_u^2 & 0 \\ 0 & \sigma_u^2 \end{pmatrix}$$

This can't be inverted:

```
Series[Inverse[θ[L] /. θ_Rules], {L, 0, 5}] // PolyForm
```

$$\begin{pmatrix} 0 & 0 \\ \frac{1}{c} & -\frac{1}{c} \end{pmatrix} + L \begin{pmatrix} 0 & 0 \\ \frac{1}{c} & -\frac{1}{c} \end{pmatrix} + L^2 \begin{pmatrix} 0 & 0 \\ \frac{1}{c} & -\frac{1}{c} \end{pmatrix} + L^3 \begin{pmatrix} 0 & 0 \\ \frac{1}{c} & -\frac{1}{c} \end{pmatrix} + L^4 \begin{pmatrix} 0 & 0 \\ \frac{1}{c} & -\frac{1}{c} \end{pmatrix} + L^5 \begin{pmatrix} 0 & 0 \\ \frac{1}{c} & -\frac{1}{c} \end{pmatrix} + \frac{\begin{pmatrix} 0 & 1 \\ 0 & -\frac{1}{c} \end{pmatrix}}{L}$$

But it gives the correct random-walk variance:

```
θ[1].Ω.Transpose[θ[1]] /. θRules /. ΩRule
```

$\{\{\sigma_u^2, \sigma_u^2\}, \{\sigma_u^2, \sigma_u^2\}\}$

(There is one random-walk component underlying both securities.)

□ *VMA representation computed from autocovariances: VMA II*

```
θII[L_] := IdentityMatrix[2] + Table[θi,j, {i, 2}, {j, 2}] L;
θII[L] // PolyForm
```

$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix} + L \begin{pmatrix} \theta_{1,1} & \theta_{1,2} \\ \theta_{2,1} & \theta_{2,2} \end{pmatrix}$

```
ΩIIRule = Ω → {{σ1^2, σ1,2}, {σ1,2, σ2^2}};
Ω /. ΩIIRule // MatrixForm
```

$\begin{pmatrix} \sigma_1^2 & \sigma_{1,2} \\ \sigma_{1,2} & \sigma_2^2 \end{pmatrix}$

```
agfII = G[θII[L], Ω /. ΩIIRule, z];
PolyForm[agfII, z, vert]
```

$\text{PolyForm}\Big[\Big\{\Big\{\frac{\theta_{1,2}\left(z\,\sigma_2^2\,\theta_{1,2} + (1 + z\,\theta_{1,1})\,\sigma_{1,2}\right)}{z} + \left(1 + \frac{\theta_{1,1}}{z}\right)\left(\sigma_1^2\,(1 + z\,\theta_{1,1}) + z\,\theta_{1,2}\,\sigma_{1,2}\right),$
$\left(1 + \frac{\theta_{2,2}}{z}\right)\left(z\,\sigma_2^2\,\theta_{1,2} + (1 + z\,\theta_{1,1})\,\sigma_{1,2}\right) + \frac{\theta_{2,1}\left(\sigma_1^2\,(1 + z\,\theta_{1,1}) + z\,\theta_{1,2}\,\sigma_{1,2}\right)}{z}\Big\},$
$\Big\{\frac{\theta_{1,2}\left(\sigma_2^2\,(1 + z\,\theta_{2,2}) + z\,\theta_{2,1}\,\sigma_{1,2}\right)}{z} + \left(1 + \frac{\theta_{1,1}}{z}\right)\left(z\,\sigma_1^2\,\theta_{2,1} + (1 + z\,\theta_{2,2})\,\sigma_{1,2}\right),$
$\left(1 + \frac{\theta_{2,2}}{z}\right)\left(\sigma_2^2\,(1 + z\,\theta_{2,2}) + z\,\theta_{2,1}\,\sigma_{1,2}\right) + \frac{\theta_{2,1}\left(z\,\sigma_1^2\,\theta_{2,1} + (1 + z\,\theta_{2,2})\,\sigma_{1,2}\right)}{z}\Big\}\Big\}, z, \text{vert}\Big]$

```
e = Union @ Flatten @ Table[MapThread[Equal,
      {Coefficient[agfI, z, k], Coefficient[agfII, z, k]}, 2], {k, 0, 1}];
e //
 TableForm
```

$0 = \sigma_2^2\,\theta_{1,2} + \theta_{1,1}\,\sigma_{1,2}$
$0 = \sigma_2^2\,\theta_{2,2} + \theta_{2,1}\,\sigma_{1,2}$
$0 = \sigma_1^2\,\theta_{1,1}\,\theta_{2,1} + \sigma_2^2\,\theta_{1,2}\,\theta_{2,2} + \sigma_{1,2} + \theta_{1,2}\,\theta_{2,1}\,\sigma_{1,2} + \theta_{1,1}\,\theta_{2,2}\,\sigma_{1,2}$
$-c^2 = \sigma_1^2\,\theta_{1,1} + \theta_{1,2}\,\sigma_{1,2}$
$\sigma_u^2 = \sigma_1^2\,\theta_{2,1} + \theta_{2,2}\,\sigma_{1,2}$
$\sigma_u^2 = \sigma_2^2 + \sigma_1^2\,\theta_{2,1}^2 + \sigma_2^2\,\theta_{2,2}^2 + 2\,\theta_{2,1}\,\theta_{2,2}\,\sigma_{1,2}$
$2\,c^2 + \sigma_u^2 = \sigma_1^2 + \sigma_1^2\,\theta_{1,1}^2 + \sigma_2^2\,\theta_{1,2}^2 + 2\,\theta_{1,1}\,\theta_{1,2}\,\sigma_{1,2}$

```
parms = {θ1,1, θ1,2, θ2,1, θ2,2, σ1^2, σ2^2, σ1,2};
```

```
solution = Solve[e, parms];
solution // Transpose // TableForm
```

$\sigma_1^2 \to \frac{c^4+3c^2\sigma_u^2+(\sigma_u^2)^2}{c^2+\sigma_u^2}$	$\sigma_1^2 \to \frac{c^4+3c^2\sigma_u^2+(\sigma_u^2)^2}{c^2+\sigma_u^2}$
$\sigma_2^2 \to \frac{c^2\sigma_u^2}{c^2+\sigma_u^2}$	$\sigma_2^2 \to \frac{c^2\sigma_u^2}{c^2+\sigma_u^2}$
$\theta_{1,2} \to \frac{c^2}{c^2+\sigma_u^2}$	$\theta_{1,2} \to \frac{c^2}{c^2+\sigma_u^2}$
$\theta_{2,2} \to -\frac{\sigma_u^2}{c^2+\sigma_u^2}$	$\theta_{2,2} \to -\frac{\sigma_u^2}{c^2+\sigma_u^2}$
$\theta_{1,1} \to -\frac{c^2}{c^2+\sigma_u^2}$	$\theta_{1,1} \to -\frac{c^2}{c^2+\sigma_u^2}$
$\theta_{2,1} \to \frac{\sigma_u^2}{c^2+\sigma_u^2}$	$\theta_{2,1} \to \frac{\sigma_u^2}{c^2+\sigma_u^2}$
$\sigma_{1,2} \to \frac{c^2\sigma_u^2}{c^2+\sigma_u^2}$	$\sigma_{1,2} \to \frac{c^2\sigma_u^2}{c^2+\sigma_u^2}$

There are two solutions, but they're the same:

```
solution[[1]] === solution[[2]]
```

```
True
```

So

```
solution = solution[[1]]
```

$$\left\{\sigma_1^2 \to \frac{c^4+3c^2\sigma_u^2+(\sigma_u^2)^2}{c^2+\sigma_u^2},\ \sigma_2^2 \to \frac{c^2\sigma_u^2}{c^2+\sigma_u^2},\ \theta_{1,2} \to \frac{c^2}{c^2+\sigma_u^2},\ \theta_{2,2} \to -\frac{\sigma_u^2}{c^2+\sigma_u^2},\ \theta_{1,1} \to -\frac{c^2}{c^2+\sigma_u^2},\ \theta_{2,1} \to \frac{\sigma_u^2}{c^2+\sigma_u^2},\ \sigma_{1,2} \to \frac{c^2\sigma_u^2}{c^2+\sigma_u^2}\right\}$$

```
θII[1] /. solution // Simplify // MatrixForm
```

$$\begin{pmatrix} \frac{\sigma_u^2}{c^2+\sigma_u^2} & \frac{c^2}{c^2+\sigma_u^2} \\ \frac{\sigma_u^2}{c^2+\sigma_u^2} & \frac{c^2}{c^2+\sigma_u^2} \end{pmatrix}$$

Consider the matrix

```
DD = (θII[L] /. L → 1 / L) * L // Simplify;
DD // MatrixForm
```

$$\begin{pmatrix} L+\theta_{1,1} & \theta_{1,2} \\ \theta_{2,1} & L+\theta_{2,2} \end{pmatrix}$$

For the VMA to be invertible, the roots of the determinental equation Det[DD]=0 must lie inside the unit circle.

We check:

```
Table[Solve[Det[DD] == 0 /. solution, L] // Flatten, {i, 2}] // MatrixForm
```

$$\begin{pmatrix} L\to 0 & L\to 1 \\ L\to 0 & L\to 1 \end{pmatrix}$$

Uh Oh. Let's try to compute the VAR by polynomial expansion:

```
Series[Inverse[θII[L]] /. solution, {L, 0, 4}] // Simplify // PolyForm
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix} + L \begin{pmatrix} \frac{c^2}{c^2+\sigma_u^2} & -\frac{c^2}{c^2+\sigma_u^2} \\ -\frac{\sigma_u^2}{c^2+\sigma_u^2} & \frac{\sigma_u^2}{c^2+\sigma_u^2} \end{pmatrix} + L^2 \begin{pmatrix} \frac{c^2}{c^2+\sigma_u^2} & -\frac{c^2}{c^2+\sigma_u^2} \\ -\frac{\sigma_u^2}{c^2+\sigma_u^2} & \frac{\sigma_u^2}{c^2+\sigma_u^2} \end{pmatrix} + L^3 \begin{pmatrix} \frac{c^2}{c^2+\sigma_u^2} & -\frac{c^2}{c^2+\sigma_u^2} \\ -\frac{\sigma_u^2}{c^2+\sigma_u^2} & \frac{\sigma_u^2}{c^2+\sigma_u^2} \end{pmatrix} + L^4 \begin{pmatrix} \frac{c^2}{c^2+\sigma_u^2} & -\frac{c^2}{c^2+\sigma_u^2} \\ -\frac{\sigma_u^2}{c^2+\sigma_u^2} & \frac{\sigma_u^2}{c^2+\sigma_u^2} \end{pmatrix}$$

No convergence here, either.

▱ *Exercise 10.1*

```
solution
```

$$\left\{\sigma_1^2 \to \frac{c^4 + 3\,c^2\,\sigma_u^2 + \left(\sigma_u^2\right)^2}{c^2+\sigma_u^2},\ \sigma_2^2 \to \frac{c^2\,\sigma_u^2}{c^2+\sigma_u^2},\ \theta_{1,2} \to \frac{c^2}{c^2+\sigma_u^2},\ \theta_{2,2} \to -\frac{\sigma_u^2}{c^2+\sigma_u^2},\ \theta_{1,1} \to -\frac{c^2}{c^2+\sigma_u^2},\ \theta_{2,1} \to \frac{\sigma_u^2}{c^2+\sigma_u^2},\ \sigma_{1,2} \to \frac{c^2\,\sigma_u^2}{c^2+\sigma_u^2}\right\}$$

```
ClearAll[VarianceDecomposition]
```

```
Ω /. ΩII_Rule
```

$$\left\{\left\{\sigma_1^2,\ \sigma_{1,2}\right\},\ \left\{\sigma_{1,2},\ \sigma_2^2\right\}\right\}$$

```
VarianceDecomposition[s_, perm_: {1, 2}] := Module[{},
   Print["Ordering of prices is: ", TableForm[{perm}]];
   F =
    Permute[CholeskyDecomposition[Permute[Ω /. ΩIIRule //. s // N, perm]], perm];
   Print["Cholesky factorization (in correct order) F=", MatrixForm[F]];
   θSum = θII[1] //. s // N;
   a = θSum.Transpose[F];
   Print["θ(1)F′=", MatrixForm[a]];
   v = a[[1]]^2;
   Print["Variance contributions:", MatrixForm[{v}]];
   tv = Total[v];
   Print["Total variance=", tv];
   Print["Information shares= ", MatrixForm[{v} / tv]];
  ];
VarianceDecomposition[Join[solution, {σu^2 → 1, c → 2}], {1, 2}]
```

Ordering of prices is: 1 2

Cholesky factorization (in correct order) $F=\begin{pmatrix} 2.40832 & 0.332182 \\ 0. & 0.830455 \end{pmatrix}$

$\theta(1)F'=\begin{pmatrix} 0.747409 & 0.664364 \\ 0.747409 & 0.664364 \end{pmatrix}$

Variance contributions: $\begin{pmatrix} 0.558621 & 0.441379 \end{pmatrix}$

Total variance=1.

Information shares= $\begin{pmatrix} 0.558621 & 0.441379 \end{pmatrix}$

```
VarianceDecomposition[Join[solution, {σu^2 → 1, c → 2}], {2, 1}]
```

Ordering of prices is: 2 1

Cholesky factorization (in correct order) $F=\begin{pmatrix} 2.23607 & 0. \\ 0.894427 & 0.894427 \end{pmatrix}$

$\theta(1)F'=\begin{pmatrix} 0.447214 & 0.894427 \\ 0.447214 & 0.894427 \end{pmatrix}$

Variance contributions: $\begin{pmatrix} 0.2 & 0.8 \end{pmatrix}$

Total variance=1.

Information shares= $\begin{pmatrix} 0.2 & 0.8 \end{pmatrix}$

□ *VECM representations.*

Here is the VAR representation for the price *levels*:

```
ϕ[L_] := Evaluate[Simplify[Inverse[θII[L]] (1 - L) /. solution] // Normal];
ϕ[L] // PolyForm
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix} + L \begin{pmatrix} -\frac{\sigma_u^2}{c^2+\sigma_u^2} & -\frac{c^2}{c^2+\sigma_u^2} \\ -\frac{\sigma_u^2}{c^2+\sigma_u^2} & -\frac{c^2}{c^2+\sigma_u^2} \end{pmatrix}$$

Verify (19.1.34) in Hamilton

```
Simplify[ϕ[1].θII[1] /. solution] // MatrixForm
```

$$\begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}$$

Rearrange $\phi(L)\, p_t = \epsilon_t$ as $p_t + \phi_1\, p_{t-1} = \epsilon_t \Longleftrightarrow p_t - p_{t-1} = (-I - \phi_1)\, p_{t-1} + \epsilon_t$. The coefficient of p_{t-1} on the r.h.s. is:

```
-IdentityMatrix[2] - Coefficient[ϕ[L], L, 1] // Together // MatrixForm
```

$$\begin{pmatrix} -\frac{c^2}{c^2+\sigma_u^2} & \frac{c^2}{c^2+\sigma_u^2} \\ \frac{\sigma_u^2}{c^2+\sigma_u^2} & -\frac{\sigma_u^2}{c^2+\sigma_u^2} \end{pmatrix}$$

This can be written as the error correction term $\gamma\, [1\ \ -1]\, p_{t-1}$ where

$$\gamma = \left\{\left\{\frac{-c^2}{\sigma_u^2 + c_1^2}\right\}, \left\{\frac{\sigma_u^2}{\sigma_u^2 + c^2}\right\}\right\};$$

```
γ.{{1, -1}} // MatrixForm
```

$$\begin{pmatrix} -\frac{c^2}{\sigma_u^2+c_1^2} & \frac{c^2}{\sigma_u^2+c_1^2} \\ \frac{\sigma_u^2}{c^2+\sigma_u^2} & -\frac{\sigma_u^2}{c^2+\sigma_u^2} \end{pmatrix}$$