LimitOrders

Empirical Market Microstructure

(2006, Oxford University Press)

Companion *Mathematica* notebook

Joel Hasbrouck

This notebook covers material related to the initial analysis of limit orders in

Chapter 12, Section 12.1

```
Text @ Style["Notebook evaluated " <> DateString["DateTime"], "Subtitle"]
```

Notebook evaluated Tuesday 5 June 2007 07:11:10

■ Initializations

```
<< Notation`
```

The following commands define symbolizations that are convenient for labeling things.

```
Symbolize[Anything_Rule]; Symbolize[Anything_Rules];
```

■ Limit order submission strategies (section 12.1)

CARA utility:

```
EURule = EU[μ_, var_] :>
  Evaluate[-CharacteristicFunction[NormalDistribution[μ, Sqrt[var]], t] /. t → I α]
```

```
EU[μ_, var_] :> -ℯ^(var α^2/2 - α μ)
```

```
CE_Rule = CE[μ_, var_] :>
  Evaluate[-Exponent[-CharacteristicFunction[NormalDistribution[μ, Sqrt[var]], t] /.
       t -> I α, E] / α // Simplify]
```

$$\text{CE}[\mu\_,\ \text{var}\_] :\to -\frac{\text{var}\ \alpha}{2} + \mu$$

Get optimal holdings via f.o.c. of certainty equivalent.

```
nOpt_Rule = First @ First @ Solve[∂_n (CE[n μ_X - n P, n^2 σ_X^2] /. CE_Rule) == 0, n]
```

$$\text{n} \to \frac{-\text{P} + \mu_X}{\alpha\ \sigma_X^2}$$

The expected utility of a limit order strategy is:

```
EU_Limit = Pr_Hit[L, λ, θ] EU[(n + 1) μ_X - L, (n + 1)^2 σ_X^2] + (1 - Pr_Hit[L, λ, θ]) EU[n μ_X, n^2 σ_X^2]
```

$$\text{EU}\left[\text{n}\ \mu_X,\ \text{n}^2\ \sigma_X^2\right]\ (1 - \text{Pr}_{\text{Hit}}[\text{L},\ \lambda,\ \theta]) + \text{EU}\left[-\text{L} + (1 + \text{n})\ \mu_X,\ (1 + \text{n})^2\ \sigma_X^2\right]\ \text{Pr}_{\text{Hit}}[\text{L},\ \lambda,\ \theta]$$

Noting that:

```
CDF[ExponentialDistribution[λ], L - θ]
```

$$\begin{cases} 1 - e^{-(\text{L}-\theta)\ \lambda} & \text{L} - \theta > 0 \end{cases}$$

We define an exponential hit probability.

```
PrHit_Rule =
 Pr_Hit[L_, λ_, θ_] :> Piecewise[{{0, L < θ}}, CDF[ExponentialDistribution[λ], L - θ]]
```

$$\text{Pr}_{\text{Hit}}[\text{L\_},\ \lambda\_,\ \theta\_] :\to \begin{cases} 0 & \text{L} < \theta \\ \text{CDF}[\text{ExponentialDistribution}[\lambda],\ \text{L} - \theta] & \text{True} \end{cases}$$

Some parameter values:

```
params = {α -> 1, σ_X^2 -> 1, μ_X -> 1, θ -> 0, λ -> 1, n -> -1};
```

```
Plot[Pr_Hit[L, λ, Θ] /. PrHit_Rule /. params, {L, 0, 10},
 AxesLabel → {L, Pr_Hit}, BaseStyle → {FontFamily → "Times"}]
```



Expected utility as a function of limit price L for initial holdings n=-2, -1, 0.

```
Plot[Evaluate[-Log[-EU_Limit] /. EU_Rule /. PrHit_Rule /. n → {-2, -1, 0} /. params],
 {L, 0, 2}, Frame → True,
 FrameLabel → {"Limit price, L", "-Log[-EULimit]", None, None},
 FrameTicks → {Automatic, None, None, None}, Background → GrayLevel[1],
 Epilog → {Text[" n=0", {0.1, -0.3}], Text[" n=-1", {0.3, -1.6}],
   Text[" n=-2", {0.8, -3.7}]}, BaseStyle → {FontFamily → "Times"}]
```



When the agent is short one share, the optimal limit price is:

```
NMaximize[EU_Limit /. EU_Rule /. PrHit_Rule /. params, {{L, 0, 1}}]
```

```
{-3.234, {L → 0.75}}
```

The probability of execution is:

```
Pr_Hit[0.75, 1, 0] /. PrHit_Rule
```

```
0.527633
```

Ask price at which customer is indifferent to doing nothing or paying the ask:

```
EU[(n + 1) μ_X - A, (n + 1)^2 σ_X^2, α] == EU[n μ_X, n^2 σ_X^2] /. EU_Rule /. params
```

```
EU[-A, 0, 1] == -ℯ^(3/2)
```

Ask price at which customer is indifferent to paying the ask or submitting a limit order at 0.75:

```
NSolve[
 EU_Limit == EU[(n + 1) μ_X - A, (n + 1)^2 σ_X^2] /. EU_Rule /. PrHit_Rule /. params /. L → 0.75, A]
```

```
— Solve::ifun :
    Inverse functions are being used by Solve, so some solutions may not
      be found; use Reduce for complete solution information. >>
```

```
{{A → 1.17372}}
```

```
Plot[Evaluate[-Log[-EU_Limit] /. EU_Rule /. PrHit_Rule /. params], {L, 0, 2},
 Frame → True, FrameLabel → {"Limit price, L", "-Log[-EULimit]", None, None},
 FrameTicks → {Automatic, None, None, None}, Background → GrayLevel[1],
 Epilog → {Text["n=0", {0.1`, -0.3`}], Text[" n=-1", {0.3`, -1.6`}],
   Text["n=-2", {0.8`, -3.7`}]}, BaseStyle → {FontFamily → "Times"}]
```




```
EU[(n + 1) μ_X - A, (n + 1)^2 σ_X^2] /. EU_Rule /. PrHit_Rule /. params /. A → 1.1
```

```
-3.00417
```