DealersAndInventories

Empirical Market Microstructure

(2006, Oxford University Press)

Companion *Mathematica* notebook

Joel Hasbrouck

This notebook covers material in Chapter 11

```
Text @ Style["Notebook evaluated " <> DateString["DateTime"], "Subtitle"]
```

Notebook evaluated Monday 4 June 2007 20:31:30

Preliminaries

```
<< Notation`
```

```
Symbolize[Anything_Rule]; Symbolize[Anything_Rules];
```

```
baseStyle = {FontFamily → "Times", FontSize → 12};
```

Poisson/exponential arrival model

```
PDF[PoissonDistribution[λ], n]
```

$$\frac{e^{-\lambda}\,\lambda^{n}}{n!}$$

```
PDF[ExponentialDistribution[α], τ]
```

$$e^{-\alpha\,\tau}\,\alpha$$

```
Mean[ExponentialDistribution[α]]
```

$\frac{1}{\alpha}$

Garman (Section 11.1.1, Exercise 11.1)

(The numbers in Exercise 11.1 were used to generate the figure.)

```
f[a_, b_, c_, x_] := a x^2 + b x + c;
```

Inverse arrival rates of sellers (who receive the bid price)

```
ps = f[.2, .2, 2, λ]
```

$2 + 0.2\lambda + 0.2\lambda^2$

```
Plot[ps, {λ, 0, 1}]
```



Inverse arrive rate of buyers (who pay the ask)

```
pb = f[.2, -1, 3, λ]
```

$3 - \lambda + 0.2\lambda^2$

The single price equilbrium

```
Plot[{ps, pb}, {λ, 0, 1}, AxesLabel → {"λ", "p"}]
```



Determination of the single-price equilibrium arrival rate:

```
λEq = λ /. Flatten[NSolve[ps == pb, λ]]
```

```
0.833333
```

```
pEq = ps /. λ → λEq
```

```
2.30556
```

(Optimal) average profit and arrival intensity:

```
sol = NMaximize[{λ (pb - ps), λ > 0}, λ]
```

```
{0.208333, {λ → 0.416667}}
```

```
λOpt = λ /. sol[[2, 1]]
```

```
0.416667
```

```
bid = ps /. λ → λOpt
```

```
2.11806
```

```
ask = pb /. λ → λOpt
```

```
2.61806
```

```
extra = {GrayLevel[0.9`], Rectangle[{0.01`, bid}, {λOpt, ask}],
   GrayLevel[0], Dashing[{0.01`}], Line[{{λOpt, 0}, {λOpt, ask}}],
   Line[{{0, bid}, {λOpt, bid}}], Line[{{0, ask}, {λOpt, ask}}],
   Line[{{λEq, 0}, {λEq, pEq}}], Line[{{0, pEq}, {λEq, pEq}}],
   Text["λ^Buy(p)", {0.3`, 2.9`}],
   Text["λ^Sell(p)", {1.05`, 2.6`}]
  };
Plot[{ps, pb}, {λ, 0, 1.2`}, PlotRange → {{0, 1.2`}, {1.8`, 3.1`}},
 Frame → True, FrameLabel → {"Arrival rate", "Price, p"},
 Background → GrayLevel[1], FrameTicks →
  {{{λOpt, "λ^Optimal "}, {λEq, "λ^Eq"}}, {{bid, Bid}, {ask, Ask}, {pEq, "p^Eq"}},
   None, None}, Epilog → extra, BaseStyle → baseStyle]
```



Risk aversion and dealer behavior (Section 11.2)

One security

The certainty equivalent in the CARA-Normal framework is given by the following rule. (There is a change of notation; the risk-aversion parameter is α.)

```
EU_Rule = EU[μ_, var_] :>
  Evaluate[-CharacteristicFunction[NormalDistribution[μ, Sqrt[var]], t] /. t → I α]
```

$$EU[\mu\_, var\_] :\to -e^{\frac{var\,\alpha^2}{2} - \alpha\,\mu}$$

```
CE_Rule = CE[μ_, var_] :>
  Evaluate[-Exponent[-CharacteristicFunction[NormalDistribution[μ, √var], t] /.
      t → I α, E] / α // Simplify]
```

$$CE[\mu\_, var\_] :\to -\frac{var\,\alpha}{2} + \mu$$

Starting with n shares, the bid at which the dealer is indifferent is:

```
B_Rule = Take @@ Flatten @@
    Solve[CE[n μ_X, n^2 σ_X^2] == CE[(n + 1) μ_X - B, (n + 1)^2 σ_X^2] /. CE_Rule, B] // Simplify
```

$$B \to \mu_X - \frac{1}{2}(1 + 2\,n)\,\alpha\,\sigma_X^2$$

with a notional price P, the optimal n is:

```
nOpt_Rule = Take @@ Flatten @@ Solve[∂_n (CE[n μ_X - n P, n^2 σ_X^2] /. CE_Rule) == 0, n]
```

$$n \to \frac{-P + \mu_X}{\alpha\,\sigma_X^2}$$

```
B_Rule /. nOpt_Rule // Simplify
```

$$B \to P - \frac{\alpha\,\sigma_X^2}{2}$$

Two securities: setting the bid with correlated risky assets

Suppose that we have two assets with payoffs $X \sim N(\mu, \Omega)$ bivariate normal. The expanded notation treats n, B, μ, and P as vectors:

```
Symbolize[n_];
Symbolize[B_];
Symbolize[P_];
Symbolize[σ_];
```

```
All_Rules = {n → {{n_1}, {n_2}}, B → {{B_1}, {B_2}}, μ → {{μ_1}, {μ_2}}, P → {{P_1}, {P_2}},
   Ω → {{σ_1^2, ρ σ_1 σ_2}, {ρ σ_1 σ_2, σ_2^2}}};
{Map[MatrixForm, All_Rules, 2]}
```

$$\left\{\left\{n \to \begin{pmatrix} n_1 \\ n_2 \end{pmatrix},\ B \to \begin{pmatrix} B_1 \\ B_2 \end{pmatrix},\ \mu \to \begin{pmatrix} \mu_1 \\ \mu_2 \end{pmatrix},\ P \to \begin{pmatrix} P_1 \\ P_2 \end{pmatrix},\ \Omega \to \begin{pmatrix} \sigma_1^2 & \rho\,\sigma_1\,\sigma_2 \\ \rho\,\sigma_1\,\sigma_2 & \sigma_2^2 \end{pmatrix}\right\}\right\}$$

Find the bid for security 1 that's consistent with the same CE before and after an execution (aquisition of one more share of security 1 at price B_1).

```
B1Rule = Take @@ Flatten @@ Solve[CE[Transpose[n].μ, Transpose[n].Ω.n] ==
      (CE[Transpose[n].μ - B1, Transpose[n].Ω.n] /. n → {{n1 + 1}, {n2}}) /.
     CERule /. AllRules, B1] // FullSimplify
```

$$B_1 \to -\frac{1}{2}\,\alpha\,\sigma_1\,(\sigma_1 + 2\,n_1\,\sigma_1 + 2\,n_2\,\rho\,\sigma_2) + \mu_1$$

At the optimum:

```
nOptRules =
  First @ Solve[0 == # & /@ (∂# Take @@ Flatten @@ (CE[Transpose[n].(μ - P), Transpose[
            n].Ω.n] /. CERule /. AllRules) & /@ {n1, n2}), {n1, n2}] // Simplify
```

$$\left\{n_1 \to \frac{P_2\,\rho\,\sigma_1 - P_1\,\sigma_2 + \sigma_2\,\mu_1 - \rho\,\sigma_1\,\mu_2}{\alpha\,\sigma_1^2\,\sigma_2 - \alpha\,\rho^2\,\sigma_1^2\,\sigma_2},\ n_2 \to \frac{P_2\,\sigma_1 - P_1\,\rho\,\sigma_2 + \rho\,\sigma_2\,\mu_1 - \sigma_1\,\mu_2}{\alpha\,\left(-1+\rho^2\right)\,\sigma_1\,\sigma_2^2}\right\}$$

```
B1Rule /. nOptRules // Simplify
```

$$B_1 \to P_1 - \frac{\alpha\,\sigma_1^2}{2}$$