LimitBookDepthII

Empirical Market Microstructure

(2006, Oxford University Press)

Companion *Mathematica* notebook

Joel Hasbrouck

This notebook covers Seppi (1997) and Sandas (2001), discussed in Chapter 13, Section 13.4

```
Text @ Style["Notebook evaluated " <> DateString["DateTime"], "Subtitle"]
```

Notebook evaluated Monday 4 June 2007 20:35:42

Initializations

```
<< Notation`
```

The following commands define symbolizations that are convenient for labeling things.

```
Symbolize[Anything_Rule]; Symbolize[Anything_Rules];
```

Equililibrium in a limit order book, Sandas (2001)

Exponential distribution (for incoming orders);

```
f_Buy[m_] := PDF[ExponentialDistribution[1/λ], m];
f_Buy[m]
```

$$\frac{e^{-\frac{m}{\lambda}}}{\lambda}$$

With $\lambda = 5$:

```
s1 = Plot[Evaluate[fBuy[m] /. λ → 5],
   {m, 10, 20}, DisplayFunction → Identity, Filling → Axis];
s2 = Plot[Evaluate[fBuy[m] /. λ → 5], {m, 0, 20}, DisplayFunction → Identity];
Show[s1, s2, PlotRange → {{0, 20}, {0, 0.2}}, DisplayFunction → $DisplayFunction,
 AxesLabel → {" m(Order size)", "Probability"},
 Ticks → {Range[0, 20, 5], Automatic},
 BaseStyle → {FontFamily → "Times"}, AxesOrigin → {0, 0}]
```



If we could condition on the size of the order, we'd impose a zero-expected profit condition for all m:

```
Solve[p - Xt - γ - α m == 0, m]
```

$$\left\{\left\{m \to \frac{p - \gamma - X_t}{\alpha}\right\}\right\}$$

A sell limit order placed at p_1

My expected profit conditional on execution is

```
Eπ1 = Simplify[∫_q^∞ (p1 - Xt - γ - α m) fBuy[m] ⅆm, Re[λ] > 0]
```

$$-e^{-\frac{q}{\lambda}} \left(\gamma + \alpha \left(q + \lambda\right) - p_1 + X_t\right)$$

I will be indifferent to adding my order to the queue at this price when $q = Q_1$ where

```
Solve[Eπ1 == 0, q] /. q → Q1
```

$$\left\{\left\{Q_1 \to \frac{-\gamma - \alpha\,\lambda + p_1 - X_t}{\alpha}\right\}\right\}$$

A sell limit order placed at p_2

```
Eπ2 = Simplify[∫_{Q1+q}^{∞} (p2 - Xt - γ - α m) fBuy[m] ⅆm, Re[λ] > 0]
```

$$-e^{-\frac{q+Q_1}{\lambda}}\left(\gamma + \alpha\,(q+\lambda) - p_2 + \alpha\,Q_1 + X_t\right)$$

Which implies:

```
Solve[Eπ2 == 0, q] /. q → Q2
```

$$\left\{\left\{Q_2 \to \frac{-\gamma - \alpha\,\lambda + p_2 - \alpha\,Q_1 - X_t}{\alpha}\right\}\right\}$$

A sell limit order placed at p_3

```
Eπ3 = Simplify[∫_{Q1+Q2+q}^{∞} (p3 - Xt - γ - α m) fBuy[m] ⅆm, Re[λ] > 0]
```

$$-e^{-\frac{q+Q_1+Q_2}{\lambda}}\left(\gamma + \alpha\,(q+\lambda) - p_3 + \alpha\,(Q_1+Q_2) + X_t\right)$$

and:

```
Solve[Eπ3 == 0, q] /. q → Q3
```

$$\left\{\left\{Q_3 \to \frac{-\gamma - \alpha\,\lambda + p_3 - \alpha\,Q_1 - \alpha\,Q_2 - X_t}{\alpha}\right\}\right\}$$

General solution

$$Q_{Rule} = Q_{k\_} :\to \mathrm{If}\left[k < \mathrm{Floor}[X_t] + 1,\ 0,\ \frac{-X_t - \gamma + p_k}{\alpha} - \lambda - \sum_{j=\mathrm{Floor}[X_t]+1}^{k-1} Q_j\right];$$

The floor of X is the largest integer less than or equal to X.

Normalize the price grid so that the tick size is unity: $p_k = k$.

$$p_{Rule} = p_{k\_} :\to k;$$

As an example, consider the numerical values:

```
InitialValues = {Xt → 0, α → .1, γ → 0, λ → 5};
```

```
QTable[ModelParms_] := Transpose[Table[
    {Qk //. (QRule /. ModelParms) /. pRule, k}, {k, Floor[Xt /. ModelParms] + 1, 5}]];
```

```
p[x_, QTable_] := Module[{k},
  k = Flatten[Position[(x < #1 &) /@ Drop[FoldList[Plus, 0, QTable[[1]]], 1], True]];
  If[Length[k] > 0, k = First[k]; QTable[[2, k]]]]
```

Here are the book schedule and value revision function:

```
QTableInitial = QTable[InitialValues]
```

```
{{5., 10., 10., 10., 10.}, {1, 2, 3, 4, 5}}
```

```
Plot[{p[x, QTableInitial], Xt + α x /. InitialValues},
 {x, 0, 40}, PlotRange → {Automatic, {0, 5}},
 PlotStyle → {{Thickness[0.005`]}, {Dashing[{0.01`}]}},
 BaseStyle → {FontFamily → "Times"}]
```



```
QTableInitial = QTable[InitialValues];
Plot[{p[x, QTableInitial], Xt + α x /. InitialValues},
 {x, 0, 14}, PlotRange → {Automatic, {0, 2.5`}},
 PlotStyle → {{Thickness[0.005`]}, {Dashing[{0.01}]}}, Frame → True,
 FrameLabel → {"Size of incoming buy order,b", " Price and EX", None, None},
 FrameTicks → {Range[0, 14, 2], Range[0, 2.5`, 0.5`], None, None},
 Background → GrayLevel[1], BaseStyle → {FontFamily → "Times"}]
```



Notice that the limit order book price schedule lies entirely above the expectation revision function. This means that if my order is the last one to execute, I realize a profit.
Suppose that the initial valuation was slightly above zero:

```
OtherValues = Join[Drop[InitialValues, 1], {Xt → .1}]
```

```
{α → 0.1, γ → 0, λ → 5, Xt → 0.1}
```

Then:

```
QTableOther = QTable[OtherValues];
Plot[{p[x, QTableOther], Xt + x α /. OtherValues},
 {x, 0, 30}, PlotRange → {Automatic, {0, 5}},
 PlotStyle → {{Thickness[0.005]}, {Dashing[{0.01`}]}}]
```



The apparent difference is that the quantities get reduced.
Does the book always start at the next higher tick above *X*? Consider:

```
OtherValues = Join[Drop[InitialValues, 1], {Xt → .8}]
```

```
{α → 0.1, γ → 0, λ → 5, Xt → 0.8}
```

In this case:

```
QTableOther = QTable[OtherValues];
Plot[{p[x, QTableOther], Xt + α x /. OtherValues},
 {x, 0, 30}, PlotRange → {Automatic, {0, 5}},
 PlotStyle → {{Thickness[0.005`]}, {Dashing[{0.01`}]}}]
```



Consider next the evolution of the book. Suppose that (starting from $\{\alpha = 0.1, \gamma = 0, \lambda = 5, X = 0\}$) we get a small order of $m = 1$. The new value of $X = \alpha m = 0.1$, so the full set of parameters is now:

```
RevisedValues = InitialValues;
RevisedValues[[1]] = Xt → (α m /. m → 1 /. InitialValues); RevisedValues
```

```
{Xt → 0.1, α → 0.1, γ → 0, λ → 5}
```

And:

```
QTableRevised = QTable[RevisedValues];
Plot[{p[x, QTableInitial], α x /. InitialValues,
  p[x, QTableRevised], Xt + α x /. RevisedValues}, {x, 0, 25},
 PlotRange → {Automatic, {0, 5}}, PlotStyle → {{Thickness[0.005`]},
   {Dashing[{0.01`}]}, {Thickness[0.01`]}, {Dashing[{0.01`}]}}]
```



Now suppose that the first order was $m = 8$, leading to revised parameters:

```
RevisedValues = InitialValues;
RevisedValues[[1]] = Xt → (α m /. m → 8 /. InitialValues); RevisedValues
```

```
{Xt → 0.8, α → 0.1, γ → 0, λ → 5}
```

and:

```
QTableRevised = QTable[RevisedValues];
Plot[{p[x, QTableInitial], α x /. InitialValues,
  p[x, QTableRevised], Xt + α x /. RevisedValues}, {x, 0, 25},
 PlotRange → {Automatic, {0, 5}}, PlotStyle → {{Thickness[0.005`]},
   {Dashing[{0.01`}]}, {Thickness[0.01`]}, {Dashing[{0.01`}]}}]
```



Originally, there were 10 shares available at a price of 2. The initial order of 8 shares left 7 shares at this price. In the new equilibrium, no additional shares were added.

Suppose we have an execution that leaves quantity q at the best price p. The book is said to "backfill" when, subsequent to the execution, additional limit orders arrive at p or better.

Conjecture 1: "backfilling" does not occur in this model.

Conjecture 2: "backfilling" might occur if we introduced event uncertainty.

■ Introduction of a dealer/specialist, Seppi (1997)

Most markets (including US equity markets) are hybrids of electronic limit order books and dealers. Dealers in this context are defined by two features: (1) They can condition their trades on the total size of the incoming order; (2) They must yield to customer orders at the same price. Seppi (1997) suggests analysis on the follow lines. To illustrate the situation, we'll take as a point of departure the ask side of the book from the Sandas model.

```
InitialValues = {Xt → 0, α → .1, γ → 0, λ → 5};
```

```
QTableInitial = QTable[InitialValues];
```

```
MoreStuff = {PointSize[0.02], Point[{8, 2}], Text["A", {8, 2.2}],
   Dashing[{.01}], Line[{{0, 0.8}, {8, 0.8}, {8, 2}}],
   Text["P(b)", {1, 1.2}], Text["E[X|b]", {12, 0.8}],
   GrayLevel[0.5], Rectangle[{5, 0.8}, {8, 1}]
  };
```

```
QTableInitial = QTable[InitialValues];
Plot[{p[x, QTableInitial], X_t + α x /. InitialValues},
 {x, 0, 14}, PlotRange → {Automatic, {0, 2.5`}}, Frame → True,
 FrameLabel → {"Size of incoming buy order, b", "Price and E[X]", None, None},
 FrameTicks → {Range[0, 14, 2], Range[0, 2.5`, 0.5`], None, None},
 Background → GrayLevel[1], PlotStyle → {{Thickness[0.005`]}, {Dashing[{0.01`}]}},
 Epilog → MoreStuff, BaseStyle → {FontFamily → "Times"}]
```