SequentialTrade

Empirical Market Microstructure

(2006, Oxford University Press)

Companion *Mathematica* notebook

Joel Hasbrouck

This notebook covers material related to sequential trade models (Chapters 5 and 6)

```
Text @ Style["Notebook evaluated " <> DateString["DateTime"], "Subtitle"]
```

Notebook evaluated Monday 4 June 2007 20:18:09

Preliminaries

```
SetDirectory[
  "c:/Active/Empirical Market Microstructure/Mathematica/Spring 2007"];
```

The `Trees` package contains various routines for working with decision trees. See the notebook `Trees Demo.nb` for documentation and examples.

```
<< Trees.m
```

Other initializations

```
<< PlotLegends`
```

```
<< Notation`
```

```
Symbolize[Anything_Rule]; Symbolize[Anything_Rules];
```

Basic one-period problem

Event tree

Functional representation of tree. (If you want to see how this is constructed, apply the rules one at a time.)

```
LabelTree =
 {V, VLo, VHi} /. x : VLo | VHi :→ {x, Inf, U} /. {VLo, Inf, a_} :→ {VLo, {Inf, S}, a} /.
   {VHi, Inf, a_} :→ {VHi, {Inf, B}, a} /. U → {U, B, S}
```

```
{V, {VLo, {Inf, S}, {U, B, S}}, {VHi, {Inf, B}, {U, B, S}}}
```

The `ShowTree` command renders a tree in a more intuitive layout. In the following diagram, a node can branch to the right and (if there is more than one alternative) to the right and down.

```
ShowTree[LabelTree]
```

```
V VLo Inf S
      U   B
          S
  VHi Inf B
      U   B
          S
```

Construct a new tree, `PrTree`, which as the same form as `LabelTree`, but in which all nodes are replaced by their conditional probabilities.

```
PrTree =
  LabelTree /. {U, B, S} → {U, 1/2, 1/2} /. S → 1 /. B → 1 /. Inf → μ /. U → 1 - μ /.
     VLo → δ /. VHi → 1 - δ /. V → 1;
BasicTree = BuildTree[LabelTree, PrTree];
ShowTree[BasicTree]
```

V 1 1	VLo δ δ	Inf μ $\delta\,\mu$	S 1 $\delta\,\mu$
		U $1-\mu$ $\delta\,(1-\mu)$	B $\frac{1}{2}$ $\frac{1}{2}\,\delta\,(1-\mu)$
			S $\frac{1}{2}$ $\frac{1}{2}\,\delta\,(1-\mu)$
	VHi $1-\delta$ $1-\delta$	Inf μ $(1-\delta)\,\mu$	B 1 $(1-\delta)\,\mu$
		U $1-\mu$ $(1-\delta)\,(1-\mu)$	B $\frac{1}{2}$ $\frac{1}{2}\,(1-\delta)\,(1-\mu)$
			S $\frac{1}{2}$ $\frac{1}{2}\,(1-\delta)\,(1-\mu)$

Conditional probabilities

Verify that terminal probabilities (buy or sell nodes) add up to 1.

```
Pr[BasicTree, S] + Pr[BasicTree, B]
```

$(1-\delta)\,(1-\mu) + \delta\,(1-\mu) + (1-\delta)\,\mu + \delta\,\mu$

... and simplify:

```
% // Simplify
```

1

The probability of B is:

```
Pr[BasicTree, B] // Simplify
```

$$\frac{1}{2}\,(1+\mu-2\,\delta\,\mu)$$

The probability of a VLo value realization AND a B is:

```
Pr[BasicTree, VLo, B] // Simplify
```

$$-\frac{1}{2}\,\delta\,(-1+\mu)$$

So using Bayes' rule, the revised δ conditional on a buy is Pr[VLo | B] =

```
δB = Pr[BasicTree, VLo, B] / Pr[BasicTree, B] // Simplify
```

$$\frac{\delta-\delta\,\mu}{1+\mu-2\,\delta\,\mu}$$

```
Simplify[∂μ δB < 0, {0 < μ < 1, 0 < δ < 1, VHi > VLo}]
```

True

```
Ask = Simplify[δB VLo + (1 - δB) VHi]
```

$$\frac{\text{VLo}\,\delta\,(-1+\mu)+\text{VHi}\,(-1+\delta)\,(1+\mu)}{-1+(-1+2\,\delta)\,\mu}$$

NOTE: Equation (5.2), p. 45 should read: $A = E[V \mid \text{Buy}] = \frac{\underline{V}\,(1-\mu)\,\delta+\overline{V}\,(1+\mu)\,(1-\delta)}{1+\mu\,(1-2\,\delta)}$

The revised δ conditional on an incoming sell is:

```
δS = Pr[BasicTree, VLo, S] / Pr[BasicTree, S] // Simplify
```

$$\frac{\delta\,(1+\mu)}{1+(-1+2\,\delta)\,\mu}$$

```
Bid = Simplify[δS VLo + (1 - δS) VHi]
```

$$\frac{\text{VHi}\,(-1+\delta)\,(-1+\mu)+\text{VLo}\,\delta\,(1+\mu)}{1+(-1+2\,\delta)\,\mu}$$

NOTE: Equation (5.6), p. 46 should read: $B = E[V \mid \text{Sell}] = \frac{\underline{V}\,(1+\mu)\,\delta+\overline{V}\,(1-\mu)\,(1-\delta)}{1-\mu\,(1-2\,\delta)}$

```
Ask - Bid // Simplify
```

$$\frac{4\,(\text{VHi}-\text{VLo})\,(-1+\delta)\,\delta\,\mu}{-1+(1-2\,\delta)^2\,\mu^2}$$

```
% /. δ → 1 / 2
```

$(\text{VHi} - \text{VLo})\ \mu$

Consider an attempt at manipulation by buying at the ask (the cash flow is $-$`Ask`) and then selling at the (revised) bid.:

```
Profits = Simplify[-Ask + (Bid /. δ → δB)]
```

$$-\frac{2\ (\text{VHi} - \text{VLo})\ (-1 + \delta)\ \delta\ \mu}{-1 + (-1 + 2\ \delta)\ \mu}$$

Test whether `Profits` can be positive, given the parameter restrictions:

```
Simplify[Profits > 0, {0 < μ < 1, 0 < δ < 1, VHi > VLo}]
```

False

There is no manipulation profit even with no informed trading in the unwind:

```
Simplify[-Ask + (Bid /. μ → 0 /. δ → δB)]
```

0

Exercise 5.1 (Different μs for different brokers)

```
Ask
```

$$\frac{\text{VLo}\ \delta\ (-1 + \mu) + \text{VHi}\ (-1 + \delta)\ (1 + \mu)}{-1 + (-1 + 2\ \delta)\ \mu}$$

```
Ask /. δ → 1 / 2 // FullSimplify
```

$$\frac{1}{2}\ (\text{VHi} + \text{VLo} + \text{VHi}\ \mu - \text{VLo}\ \mu)$$

```
Profits_b = (Ask /. μ → μ_Other) - (Ask /. μ → μ_b) /. δ → 1 / 2 // FullSimplify
```

$$-\frac{1}{2}\ (\text{VHi} - \text{VLo})\ (\mu_b - \mu_{\text{Other}})$$

```
Simplify[Profits_b > 0, {0 < μ_b < 1, 0 < μ_Other < 1, μ_b < μ_Other, VHi > VLo, 0 < δ < 1}]
```

True

Exercise 5.2 (Informed trading only in low state)

```
LabelTree = {V, {VLo, {In, S}, {Un, B, S}}, {VHi, {Un, B, S}}};
LabelTree // ShowTree
```

```
V VLo In S
      Un B
         S
  VHi Un B
         S
```

```
PrTree = LabelTree /. {VLo, {In, S}, a_} ⧴ {VLo, {In, 1}, a} /. B | S → 1 / 2 /. In → μ /.
       {VHi, {Un, a__}} ⧴ {VHi, {1, a}} /. Un → 1 - μ /. VLo → δ /. VHi → 1 - δ /. V → 1;
ExTree = BuildTree[LabelTree, PrTree];
ShowTree[ExTree]
```

V 1 1	VLo δ δ	In μ $\delta\,\mu$	S 1 $\delta\,\mu$
		Un $1-\mu$ $\delta\,(1-\mu)$	B $\frac{1}{2}$ $\frac{1}{2}\,\delta\,(1-\mu)$
			S $\frac{1}{2}$ $\frac{1}{2}\,\delta\,(1-\mu)$
	VHi $1-\delta$ $1-\delta$	Un 1 $1-\delta$	B $\frac{1}{2}$ $\frac{1-\delta}{2}$
			S $\frac{1}{2}$ $\frac{1-\delta}{2}$

Verify that terminal node total probabilities add up to one:

```
Pr[ExTree, B] + Pr[ExTree, S] // Simplify
```

```
1
```

Conditional on a buy, the revised δ is:

```
δB = Pr[ExTree, VLo, B] / Pr[ExTree, B] // Simplify
```

$$\frac{\delta\,(-1+\mu)}{-1+\delta\,\mu}$$

Conditional on a sell ...

```
δS = Pr[ExTree, VLo, S] / Pr[ExTree, S] // Simplify
```

$$\frac{\delta\,(1+\mu)}{1+\delta\,\mu}$$

```
Ask = Simplify[δB VLo + (1 - δB) VHi]
```

$$\frac{\text{VHi}\,(-1+\delta)+\text{VLo}\,\delta\,(-1+\mu)}{-1+\delta\,\mu}$$

```
Bid = Simplify[δS VLo + (1 - δS) VHi]
```

$$\frac{\text{VHi}-\text{VHi}\,\delta+\text{VLo}\,\delta\,(1+\mu)}{1+\delta\,\mu}$$

```
Simplify[δS /. δ → δB]
```

$$\frac{\delta\,(-1+\mu)\,(1+\mu)}{-1+\delta\,\mu^2}$$

Manipulation by buying at the ask and selling at the revised bid? Nope.

```
Simplify[(-Ask + Bid /. δ → δB) > 0, {0 < μ < 1, 0 < δ < 1, VHi > VLo}]
```

```
False
```

Manipulation by selling at the bid and buying at the revised ask? Nope.

```
Simplify[(Bid - Ask /. δ → δS) > 0, {0 < μ < 1, 0 < δ < 1, VHi > VLo}]
```

```
False
```

Exercise 5.3 (Informed traders get a signal)

```
LabelTree =
 {V,
  {VLo,
   {In, {SLo, S}, {SHi, B}},
   {Un, B, S}},
  {VHi,
   {In, {SLo, S}, {SHi, B}},
   {Un, B, S}}
 }; LabelTree // ShowTree
```

```
V VLo In SLo S
         SHi B
      Un B
         S
  VHi In SLo S
         SHi B
      Un B
         S
```

```
PrTree =
  LabelTree /. {VLo, {In, {SLo, S}, {SHi, B}}, a_} :> {VLo, {In, {γ, 1}, {1 - γ, 1}},
            a} /. {VHi, {In, {SLo, S}, {SHi, B}}, a_} :>
          {VHi, {In, {1 - γ, 1}, {γ, 1}}, a} /. B | S → 1 / 2 /.
       In → μ /. Un → 1 - μ /. VLo → δ /. VHi → 1 - δ /. V → 1;
ExTree = BuildTree[LabelTree, PrTree];
ShowTree[ExTree]
```

V 1 1	VLo δ δ	In μ $\delta\,\mu$	SLo γ $\gamma\,\delta\,\mu$	S 1 $\gamma\,\delta\,\mu$
			SHi $1-\gamma$ $(1-\gamma)\,\delta\,\mu$	B 1 $(1-\gamma)\,\delta\,\mu$
		Un $1-\mu$ $\delta\,(1-\mu)$	B $\frac{1}{2}$ $\frac{1}{2}\,\delta\,(1-\mu)$	
			S $\frac{1}{2}$ $\frac{1}{2}\,\delta\,(1-\mu)$	
	VHi $1-\delta$ $1-\delta$	In μ $(1-\delta)\,\mu$	SLo $1-\gamma$ $(1-\gamma)\,(1-\delta)\,\mu$	S 1 $(1-\gamma)\,(1-\delta)\,\mu$
			SHi γ $\gamma\,(1-\delta)\,\mu$	B 1 $\gamma\,(1-\delta)\,\mu$
		Un $1-\mu$ $(1-\delta)\,(1-\mu)$	B $\frac{1}{2}$ $\frac{1}{2}\,(1-\delta)\,(1-\mu)$	
			S $\frac{1}{2}$ $\frac{1}{2}\,(1-\delta)\,(1-\mu)$	

```
Pr[ExTree, B] + Pr[ExTree, S] // Simplify
```

1

```
δB = Pr[ExTree, VLo, B] / Pr[ExTree, B] // Simplify
```

$$\frac{\delta\,(-1+(-1+2\,\gamma)\,\mu)}{-1+(-1+2\,\gamma)\,(-1+2\,\delta)\,\mu}$$

```
Values = {γ → .1, μ → .2, δ → .3};
```

```
δB /. Values
```

```
0.371795
```

```
Ask = Simplify[δB VLo + (1 - δB) VHi]
```

$$\frac{\text{VLo}\,\delta\,(-1+(-1+2\gamma)\,\mu)+\text{VHi}\,(-1+\delta)\,(1+(-1+2\gamma)\,\mu)}{-1+(-1+2\gamma)\,(-1+2\delta)\,\mu}$$

```
δS = Pr[ExTree, VLo, S] / Pr[ExTree, S] // Simplify
```

$$\frac{\delta\,(1+(-1+2\gamma)\,\mu)}{1+(-1+2\gamma)\,(-1+2\delta)\,\mu}$$

```
Bid1 = Simplify[δS VLo + (1 - δS) VHi]
```

$$\frac{\text{VHi}\,(-1+\delta)\,(-1+(-1+2\gamma)\,\mu)+\text{VLo}\,\delta\,(1+(-1+2\gamma)\,\mu)}{1+(-1+2\gamma)\,(-1+2\delta)\,\mu}$$

```
Simplify[δS /. δ → δB]
```

$$\delta$$

Sequences of buys and sells

δCond returns δ_k as a function of δ_{k-1} and the kth trade (B or S):

$$\delta\text{Cond}_{\text{Rule}} = \delta\text{Cond}[\delta\_,\ x\_] :\to \text{If}\left[x === B,\ \frac{\delta-\delta\,\mu}{1+\mu-2\,\delta\,\mu},\ \frac{\delta\,(1+\mu)}{1+(-1+2\,\delta)\,\mu}\right];$$

This can be used to obtain the revised δ for the first trade ...

$$\delta\text{Cond}[\delta_0,\ S]\ /.\ \delta\text{Cond}_{\text{Rule}}$$

$$\frac{(1+\mu)\,\delta_0}{1+\mu\,(-1+2\,\delta_0)}$$

... or a sequence of trades:

$$\text{FoldList}[\delta\text{Cond},\ \delta_0,\ \{B,\ S\}]$$

$$\{\delta_0,\ \delta\text{Cond}[\delta_0,\ B],\ \delta\text{Cond}[\delta\text{Cond}[\delta_0,\ B],\ S]\}$$

```
% //. δCond_Rule // Simplify
```

$$\left\{\delta_0, \frac{(-1+\mu)\,\delta_0}{-1-\mu+2\,\mu\,\delta_0}, \delta_0\right\}$$

```
Simplify[∂_μ (δCond[δ, S] /. δCond_Rule) > 0, {0 < δ < 1, 0 < μ < 1}]
```

True

```
Simplify[∂_μ (δCond[δ, B] /. δCond_Rule) < 0, {0 < δ < 1, 0 < μ < 1}]
```

True

```
FoldList[δCond, δ0, {B, B, S, S}]
```

{δ0, δCond[δ0, B], δCond[δCond[δ0, B], B],
 δCond[δCond[δCond[δ0, B], B], S], δCond[δCond[δCond[δCond[δ0, B], B], S], S]}

```
% //. δCond_Rule // Simplify
```

$$\left\{\delta_0, \frac{(-1+\mu)\,\delta_0}{-1-\mu+2\,\mu\,\delta_0}, \frac{(-1+\mu)^2\,\delta_0}{(1+\mu)^2-4\,\mu\,\delta_0}, \frac{(-1+\mu)\,\delta_0}{-1-\mu+2\,\mu\,\delta_0}, \delta_0\right\}$$

Exercise 5.4 (Offsetting trades)

A sell followed by a buy gives ... (see the analysis of the original problem, above).

```
FoldList[δCond, δ0, {S, B}] //. δCond_Rule // Simplify
```

$$\left\{\delta_0, \frac{(1+\mu)\,\delta_0}{1-\mu+2\,\mu\,\delta_0}, \delta_0\right\}$$

"Price Impact" (section 5.6)

Consider two sells followed by a buy:

```
FoldList[δCond, δ0, {S, S, B}] //. δCond_Rule // Simplify
```

$$\left\{\delta_0, \frac{(1+\mu)\,\delta_0}{1-\mu+2\,\mu\,\delta_0}, \frac{(1+\mu)^2\,\delta_0}{(-1+\mu)^2+4\,\mu\,\delta_0}, \frac{(1+\mu)\,\delta_0}{1-\mu+2\,\mu\,\delta_0}\right\}$$

Using `Fold` instead of `FoldList` returns the last element:

```
δMM = (Fold[δCond, δ0, {S, S, B}] //. δCondRule // Simplify)
```

$$\frac{(1+\mu)\,\delta_0}{1-\mu+2\,\mu\,\delta_0}$$

```
ParameterRestrictions = {0 < μ < 1, 0 < δ0 < 1, VHi > VLo};
```

```
Simplify[δMM > δ0, ParameterRestrictions]
```

True

So the MM has a more pessimistic view of value than the order-informed trader, even after the order-informed trader has bought.

```
Ask = Simplify[δMM VLo + (1 - δMM) VHi]
```

$$\frac{\mathrm{VHi} - \mathrm{VHi}\,\mu + (\mathrm{VHi}\,(-1+\mu) + \mathrm{VLo}\,(1+\mu))\,\delta_0}{1-\mu+2\,\mu\,\delta_0}$$

```
EV = δ0 VLo + (1 - δ0) VHi
```

$\mathrm{VHi}\,(1-\delta_0) + \mathrm{VLo}\,\delta_0$

The order-informed trader's expected profits are:

```
Simplify[EV - Ask]
```

$$-\frac{2\,(\mathrm{VHi}-\mathrm{VLo})\,\mu\,(-1+\delta_0)\,\delta_0}{1-\mu+2\,\mu\,\delta_0}$$

The expected profits are positive

```
Simplify[EV - Ask > 0, ParameterRestrictions]
```

True

```
Ask /. δ0 → 1 / 2 /. VLo → 0 // Simplify
```

$$\frac{1}{2}\,(\mathrm{VHi} - \mathrm{VHi}\,\mu)$$

■ The distribution of buys and sells (Chapter 6)

```
PDF[BinomialDistribution[n, p], nBuys]
```

$(1-p)^{n-\mathrm{nBuys}}\, p^{\mathrm{nBuys}}\, \mathrm{Binomial}[n, \mathrm{nBuys}]$

```
Clear[p, a, b, n, δ]
```

```
PN = (δ PDF[BinomialDistribution[n, p], b] /. p → (1 - μ)/2) +
   ((1 - δ) PDF[BinomialDistribution[n, p], b] /. p → (1 - μ)/2 + μ);
```

```
PN
```

$$2^{-b}\,\delta\left(1+\frac{1}{2}(-1+\mu)\right)^{-b+n}(1-\mu)^{b}\,\mathrm{Binomial}[n,b]+(1-\delta)\left(1+\frac{1}{2}(-1+\mu)-\mu\right)^{-b+n}\left(\frac{1-\mu}{2}+\mu\right)^{b}\mathrm{Binomial}[n,b]$$

```
a = Table[{b, PN /. n → 50 /. δ → 1/2}, {b, 0, 50}];
bPlot = N[{a /. μ → 0, a /. μ → .1, a /. μ → .5}];
```

```
lineStyles =
  Table[{GrayLevel[0], Dashing[.01 r], AbsoluteThickness[.5 r]}, {r, 0, 2}];
```

```
a = ListPlot[bPlot, PlotJoined → True,
  Frame → True, FrameTicks → {Automatic, Automatic, None, None},
  FrameLabel → {"Number of buys in fifty trades, b)", "Probability"},
  PlotStyle → lineStyles, BaseStyle → {FontFamily → "Times"}]
```




```
legendLines = Table[Graphics[{GrayLevel[0], Dashing[.1 r],
     AbsoluteThickness[.5 r], Line[{{0, 0}, {2, 0}}]}], {r, 0, 2}];
```

```
legendText = Style[#, Black, FontFamily → "Times"] & /@ {0, 0.1, 0.5};
```

```
legendEntries = Transpose[{legendLines, legendText}];
```

```
ShowLegend[a, {legendEntries, LegendPosition → {1, -.4},
  LegendSize → 0.6, LegendShadow → None, LegendLabel → μ}]
```




```
PDF[PoissonDistribution[μ], n]
```

$$\frac{e^{-\mu}\,\mu^{n}}{n!}$$

```
Clear[f];
fMixDefinition =
 fMix :→ (1 - α) f[ϵ, B] f[ϵ, S] + α δ (f[ϵ + μ, B] f[ϵ, S]) + α (1 - δ) f[ϵ + μ, S] f[ϵ, B]
```

```
fMix :→ (1 - α) f[ϵ, B] f[ϵ, S] + α δ (f[ϵ + μ, B] f[ϵ, S]) + α (1 - δ) f[ϵ + μ, S] f[ϵ, B]
```

```
p1 = ListPlot[{Table[{n, N[PDF[PoissonDistribution[10], n]]}, {n, 0, 20}]},
   DisplayFunction → Identity];
```

```
p2 = Plot[PDF[NormalDistribution[10, √10], x],
   {x, 0, 20}, DisplayFunction → Identity];
```

```
Show[p1, p2, BaseStyle → {FontFamily → "Times"}]
```



Mixture aspects of Easley, Kiefer, O'Hara and Paperman (section 6.2)

$$\texttt{fMix /. fMixDefinition /. } \delta \to \frac{1}{2}$$

$$(1-\alpha)\, f[\epsilon, B]\, f[\epsilon, S] + \frac{1}{2}\alpha\, f[\epsilon, S]\, f[\epsilon+\mu, B] + \frac{1}{2}\alpha\, f[\epsilon, B]\, f[\epsilon+\mu, S]$$

$$\texttt{fMixApprox =}$$
$$\texttt{fMix /. fMixDefinition /. f[}\lambda\texttt{\_, x\_]} :\to \texttt{PDF}\left[\texttt{NormalDistribution}\left[\lambda, \sqrt{\lambda}\right], x\right] \texttt{ /. } \delta \to \frac{1}{2}$$

$$\frac{e^{-\frac{(B-\epsilon)^2}{2\epsilon}-\frac{(S-\epsilon)^2}{2\epsilon}}(1-\alpha)}{2\pi\epsilon} + \frac{e^{-\frac{(S-\epsilon)^2}{2\epsilon}-\frac{(B-\epsilon-\mu)^2}{2(\epsilon+\mu)}}\alpha}{4\pi\sqrt{\epsilon}\sqrt{\epsilon+\mu}} + \frac{e^{-\frac{(B-\epsilon)^2}{2\epsilon}-\frac{(S-\epsilon-\mu)^2}{2(\epsilon+\mu)}}\alpha}{4\pi\sqrt{\epsilon}\sqrt{\epsilon+\mu}}$$

```
TestValues = {α → .4, ϵ → 10, μ → 10}
```

```
{α → 0.4, ϵ → 10, μ → 10}
```

```
p1 = ContourPlot[N[Log[fMixApprox /. TestValues]], {S, 0, 30}, {B, 0, 30},
   Contours → 20, FrameLabel → {"Number of buys", "Number of sells", None, None},
   DisplayFunction → Identity, BaseStyle → {FontFamily → "Times"}];
```

```
p2 = Plot3D[N[fMixApprox /. TestValues], {S, 0, 30},
   {B, 0, 30}, BaseStyle → {FontFamily → "Times", FontSize → 12},
   PlotPoints → 50, PlotRange → All, ViewPoint → 2 {1, 0.9`, 0.4`},
   AxesLabel → {"Sells", "Buys", None}, Ticks → None, DisplayFunction → Identity];
```

```
Show[GraphicsRow[{p1, p2}]]
```



□ *Supplementary results*

The "base" density here is a bivariate normal centered at ten buys and ten sells. The stretching of the density along the *B* and *S* axes reflects the higher arrival rates on information days. The canyon between the two lobes occurs because if an information event occurs, good and bad news are mutually exclusive. The density above is representative. Changes in the parameter values can dramatically distort the picture, as the following examples show.

```
g[μ_, α_, ϵ_] := Evaluate[Log[fMixApprox]];
gg[μ_, α_, ϵ_] := ContourPlot[g[μ, α, ϵ],
   {S, 0, 50}, {B, 0, 50}, Contours → 30, DisplayFunction → Identity,
   PlotLabel → "μ=" <> ToString[μ] <> " α=" <> ToString[NumberForm[α, {3, 2}]] <>
     " ϵ=" <> ToString[ϵ] <> " PIN=" <> ToString[NumberForm[(α μ)/(α μ + 2 ϵ), {3, 2}]],
   BaseStyle → {FontFamily → "Times", FontSize → 12}];
```

```
Show[GraphicsGrid[
  {{gg[10, 0.1`, 10], gg[10, 0.2`, 10]}, {gg[10, 0.5`, 10], gg[10, 0.99`, 10]}}]]
```



When the characteristics of a distribution are strongly dependent on parameter values, different sample distributions will imply different parameter values, i.e., the data are likely to informative in estimating the parameters with precision.

PIN, however, is a derived, summary quantity. Suppose that we investigate a set of distributions where *PIN* is held constant at 0.10.

$$\alpha\text{Rule} = \text{Flatten}\left[\text{Solve}\left[\text{PIN} == \frac{\alpha\,\mu}{\alpha\,\mu + 2\,\epsilon} \;/.\; \text{PIN} \to .1 \;/.\; \epsilon \to 10,\ \alpha\right]\right];$$

```
ggg[μ_] := gg[μ, 2.2222222`/μ, 10];
Show[GraphicsGrid[{{ggg[10], ggg[9]}, {ggg[8], ggg[7]}}]]
```




```
Solve[PIN == αμ/(αμ + 2ϵ) /. PIN → .4 /. ϵ → 10 /. α → 1, μ]
```

{{μ → 13.3333}}

```
Solve[PIN == αμ/(αμ + 2ϵ) /. PIN → .4 /. ϵ → 10 /. α → .9, μ]
```

{{μ → 14.8148}}

```
ggg[μ_] := gg[μ, 2.2222222`/μ, 10];
Show[GraphicsGrid[{{gg[13.3`, 1, 10], gg[14.8`, 0.9`, 10]}}]]
```