LimitBookDepthI

Empirical Market Microstructure

(2006, Oxford University Press)

Companion *Mathematica* notebook

Joel Hasbrouck

This notebook covers Glosten (1989, 1994) discussed in Chapter 13, Sections 13.1-13.3

```
Text @ Style["Notebook evaluated " <> DateString["DateTime"], "Subtitle"]
```

Notebook evaluated Monday 4 June 2007 20:35:42

Preliminaries

```
SetDirectory[
  "c:/Active/Empirical Market Microstructure/Mathematica/Spring 2007"];
```

```
<< MVN.m
```

```
<< Notation`
```

The following commands define symbolizations that are convenient for labeling things.

```
Symbolize[Anything_Rule]; Symbolize[Anything_Rules]
```

The following command allows the "expectation" operator to be entered as ESC E ESC.

```
AddInputAlias[ℰ, "E"]
```

Rules related to normal distribution

ϕ[x], Φ[X], etc.

ϕ[x] and Φ[x] are often used to denote the density and distribution functions for the standard normal distribution.

```
NormalRules = {
   ϕ[x_] :→ PDF[NormalDistribution[0, 1], x],
   Φ[x_] :→ CDF[NormalDistribution[0, 1], x],
   ϕ[x_, μ_, var_] :→ PDF[NormalDistribution[μ, √var], x],
   Φ[x_, μ_, var_] :→ CDF[NormalDistribution[μ, √var], x]};
```

Expectations of truncated normal variates

First consider E[x | x ≥ xLow] where x is normally distributed. In the problems considered here, x will always be an element of a vector that is distributed as a multivariate normal denoted MVN. (See the MVN.m package and the MVN Demo.nb notebook.) The truncated expectation is:

```
GetMeanTrunc[m_MVN, x_, xLow_] := Module[{d, μ, v, α, λ},
  d = MakeLinearForm[m, {x}];
  μ = GetMean[d];
  v = GetVariance[d];
  α = (xLow - μ) / √v;
  λ = ϕ[α] / (1 - Φ[α]);
  μ + √v λ]
```

where m is an MVN that contains variable x. Example:

```
mSample = MVN[{μ1, μ2}, {{σ1^2, σ12}, {σ12, σ2^2}}, {x1, x2}]
```

$$\begin{pmatrix} x_1 \\ x_2 \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} \mu_1 \\ \mu_2 \end{pmatrix}, \begin{pmatrix} \sigma_1^2 & \sigma_{12} \\ \sigma_{12} & \sigma_2^2 \end{pmatrix}\right)$$

Then $E[x_1 \mid x_1 \geq x_{Low}]$ is:

```
GetMeanTrunc[mSample, x1, xLow]
```

$$\mu_1 + \frac{\sqrt{\sigma_1^2}\ \phi\left[\frac{x_{Low}-\mu_1}{\sqrt{\sigma_1^2}}\right]}{1-\Phi\left[\frac{x_{Low}-\mu_1}{\sqrt{\sigma_1^2}}\right]}$$

```
GetMeanTrunc[m_MVN, y_, x_, xLow_] := Module[{d},
   d = MVNConditional[m, y, x];
   GetMean[d] /. x → GetMeanTrunc[m, x, xLow]];
```

```
GetMeanTrunc[StandardMVN[2], z1, 0] /. NormalRules
```

$$\sqrt{\frac{2}{\pi}}$$

```
GetMeanTrunc[MVN[2, 9, z], z, 1] /. NormalRules // N
```

3.79547

```
GetMeanTrunc[MVN[{3, 2}, {{8, 3}, {2, 3}}, {x, y}], y, x, 2] /. NormalRules // N
```

2.62289

Setup common to all market variants.

Customer's utility

Certainty equivalent with CARA utility and normally-distributed terminal wealth:

```
CERule = CE[μ_, var_] :→
  Evaluate[-Exponent[-CharacteristicFunction[NormalDistribution[μ, √var], t] /.
       t → I ρ, E] / ρ // Simplify]
```

$$CE[\mu\_,\ var\_] \mapsto \mu - \frac{var\ \rho}{2}$$

The customer is endowed with n shares; she purchases q shares, paying R[q] (the dealer's revenue). Her certainty equivalent wealth is:

```
ce = CE[-R[q] + (n + q) εCond, (n + q)^2 VarCond] /. CERule
```

$$(n+q)\ \mathcal{E}Cond - \frac{1}{2}\ (n+q)^2\ VarCond\ \rho - R[q]$$

where $\mathcal{E}$Cond and VarCond are the mean and variance of the security payoff, conditional on the customer's signal.

The first-order condition is:

```
Solve[(∂_q ce == 0), R′[q]] // Simplify
```

$$\{\{R'[q] \to \mathcal{E}\text{Cond} - (n + q)\ \text{VarCond}\ \rho\}\}$$

Defining the l.h.s. as the marginal revenue M we have:

```
M_Rule = Take @@ Take @@ % /. R′[q] → M // Simplify
```

$$M \to \mathcal{E}\text{Cond} - (n + q)\ \text{VarCond}\ \rho$$

Given q, knowing M is equivalent to knowing ω where

```
ω_Rule = ω → ℰCond - n ρ VarCond
```

$$\omega \to \mathcal{E}\text{Cond} - n\ \text{VarCond}\ \rho$$

Let x, ϵ, and n denote (respectively) the secuity payoff, the signal observation error, and the customer's endowment. Then:

```
d1 = MVN[{μ_x, 0, 0}, {{σ_x^2, 0, 0}, {0, σ_ϵ^2, 0}, {0, 0, σ_n^2}}, {x, ϵ, n}]
```

$$\begin{pmatrix} x \\ \epsilon \\ n \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} \mu_x \\ 0 \\ 0 \end{pmatrix}, \begin{pmatrix} \sigma_x^2 & 0 & 0 \\ 0 & \sigma_\epsilon^2 & 0 \\ 0 & 0 & \sigma_n^2 \end{pmatrix}\right)$$

Since the signal is $s = x + \epsilon$, we may write:

```
MakeLinearForm[d1, {x, x + ϵ, n}]
```

$$\begin{pmatrix} x \\ x + \epsilon \\ n \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} \mu_x \\ \mu_x \\ 0 \end{pmatrix}, \begin{pmatrix} \sigma_x^2 & \sigma_x^2 & 0 \\ \sigma_x^2 & \sigma_x^2 + \sigma_\epsilon^2 & 0 \\ 0 & 0 & \sigma_n^2 \end{pmatrix}\right)$$

Now we simply label $x + \epsilon$ as s:

```
d2 = SetLabel[%, {x, s, n}]
```

$$\begin{pmatrix} x \\ s \\ n \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} \mu_x \\ \mu_x \\ 0 \end{pmatrix}, \begin{pmatrix} \sigma_x^2 & \sigma_x^2 & 0 \\ \sigma_x^2 & \sigma_x^2 + \sigma_\epsilon^2 & 0 \\ 0 & 0 & \sigma_n^2 \end{pmatrix}\right)$$

The conditional distribution of $x \mid s$ is:

```
d2Cond = MVNConditional[d2, x, s] // Simplify
```

$$x \sim \mathcal{N}\left(\frac{s\,\sigma_x^2 + \mu_x\,\sigma_\epsilon^2}{\sigma_x^2 + \sigma_\epsilon^2}, \frac{\sigma_x^2\,\sigma_\epsilon^2}{\sigma_x^2 + \sigma_\epsilon^2}\right)$$

Substituting the conditional mean and variance into the expression for ω gives:

```
ω2Rule = ωRule /. ℰCond → GetMean[d2Cond] /. VarCond → GetVariance[d2Cond] // Simplify
```

$$\omega \to \frac{\mu_x\,\sigma_\epsilon^2 + \sigma_x^2\left(s - n\,\rho\,\sigma_\epsilon^2\right)}{\sigma_x^2 + \sigma_\epsilon^2}$$

Then

```
MakeLinearForm[d2, {x, s, n, ω /. ω2Rule}] // Simplify
```

$$\begin{pmatrix} x \\ s \\ n \\ \frac{\mu_x\,\sigma_\epsilon^2 + \sigma_x^2\left(s - n\,\rho\,\sigma_\epsilon^2\right)}{\sigma_x^2 + \sigma_\epsilon^2} \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} \mu_x \\ \mu_x \\ 0 \\ \mu_x \end{pmatrix}, \begin{pmatrix} \sigma_x^2 & \sigma_x^2 & 0 & \frac{\sigma_x^4}{\sigma_x^2+\sigma_\epsilon^2} \\ \sigma_x^2 & \sigma_x^2 + \sigma_\epsilon^2 & 0 & \sigma_x^2 \\ 0 & 0 & \sigma_n^2 & -\frac{\rho\,\sigma_n^2\,\sigma_x^2\,\sigma_\epsilon^2}{\sigma_x^2+\sigma_\epsilon^2} \\ \frac{\sigma_x^4}{\sigma_x^2+\sigma_\epsilon^2} & \sigma_x^2 & -\frac{\rho\,\sigma_n^2\,\sigma_x^2\,\sigma_\epsilon^2}{\sigma_x^2+\sigma_\epsilon^2} & \frac{\sigma_x^4\left(\sigma_x^2+\sigma_\epsilon^2+\rho^2\,\sigma_n^2\,\sigma_\epsilon^4\right)}{\left(\sigma_x^2+\sigma_\epsilon^2\right)^2} \end{pmatrix}\right)$$

and relabeling the last element as ω gives:

```
d3 = SetLabel[%, {x, s, n, ω}]
```

$$\begin{pmatrix} x \\ s \\ n \\ \omega \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} \mu_x \\ \mu_x \\ 0 \\ \mu_x \end{pmatrix}, \begin{pmatrix} \sigma_x^2 & \sigma_x^2 & 0 & \frac{\sigma_x^4}{\sigma_x^2+\sigma_\epsilon^2} \\ \sigma_x^2 & \sigma_x^2 + \sigma_\epsilon^2 & 0 & \sigma_x^2 \\ 0 & 0 & \sigma_n^2 & -\frac{\rho\,\sigma_n^2\,\sigma_x^2\,\sigma_\epsilon^2}{\sigma_x^2+\sigma_\epsilon^2} \\ \frac{\sigma_x^4}{\sigma_x^2+\sigma_\epsilon^2} & \sigma_x^2 & -\frac{\rho\,\sigma_n^2\,\sigma_x^2\,\sigma_\epsilon^2}{\sigma_x^2+\sigma_\epsilon^2} & \frac{\sigma_x^4\left(\sigma_x^2+\sigma_\epsilon^2+\rho^2\,\sigma_n^2\,\sigma_\epsilon^4\right)}{\left(\sigma_x^2+\sigma_\epsilon^2\right)^2} \end{pmatrix}\right)$$

The liquidity suppliers will try to estimate $x \mid \omega$. The relevant conditional distribution is:

```
d3Cond = MVNConditional[d3, x, ω]
```

$$x \sim \mathcal{N}\left(\mu_x + \frac{(\omega - \mu_x)\left(\sigma_x^2 + \sigma_\epsilon^2\right)}{\sigma_x^2 + \sigma_\epsilon^2 + \rho^2\,\sigma_n^2\,\sigma_\epsilon^4}, \sigma_x^2 - \frac{\sigma_x^4}{\sigma_x^2 + \sigma_\epsilon^2 + \rho^2\,\sigma_n^2\,\sigma_\epsilon^4}\right)$$

▫ $E[x \mid M = m]$:

The expectation of x given the that the marginal revenue is exactly equal to m is

```
ℰPoint[m_] :=
 Evaluate[GetMean[d3Cond] /. ω → m + q ρ GetVariance[d2Cond] // Simplify]
```

```
ℰPoint[m]
```

$$\frac{\sigma_x^2\left(m+q\,\rho\,\sigma_\epsilon^2\right)+\sigma_\epsilon^2\left(m+\rho^2\,\mu_x\,\sigma_n^2\,\sigma_\epsilon^2\right)}{\sigma_x^2+\sigma_\epsilon^2+\rho^2\,\sigma_n^2\,\sigma_\epsilon^4}$$

$E[x \mid M \geq m]$

The expectation of x given that the marginal revenue is greater than or equal to m is the truncated ("upper tail") expectation.

```
ute = GetMeanTrunc[d3, x, ω, ωLow] /. ωLow → m + q ρ GetVariance[d2Cond] // Simplify
```

$$\mu_x-\frac{\sigma_x^4\,\phi\left[\frac{m-\mu_x+\frac{q\,\rho\,\sigma_x^2\,\sigma_\epsilon^2}{\sigma_x^2+\sigma_\epsilon^2}}{\sqrt{\frac{\sigma_x^4\left(\sigma_x^2+\sigma_\epsilon^2+\rho^2\,\sigma_n^2\,\sigma_\epsilon^4\right)}{\left(\sigma_x^2+\sigma_\epsilon^2\right)^2}}}\right]}{\left(\sigma_x^2+\sigma_\epsilon^2\right)\sqrt{\frac{\sigma_x^4\left(\sigma_x^2+\sigma_\epsilon^2+\rho^2\,\sigma_n^2\,\sigma_\epsilon^4\right)}{\left(\sigma_x^2+\sigma_\epsilon^2\right)^2}}\left(-1+\Phi\left[\frac{m-\mu_x+\frac{q\,\rho\,\sigma_x^2\,\sigma_\epsilon^2}{\sigma_x^2+\sigma_\epsilon^2}}{\sqrt{\frac{\sigma_x^4\left(\sigma_x^2+\sigma_\epsilon^2+\rho^2\,\sigma_n^2\,\sigma_\epsilon^4\right)}{\left(\sigma_x^2+\sigma_\epsilon^2\right)^2}}}\right]\right)}$$

where ϕ and Φ are the density and c.d.f. of the standard normal distibution. Writing this as a function of m:

```
ℰTail[m_] := Evaluate[ute /. Normal_Rules]
```

Common parameters used to generate (most) graphs

```
CommonParams = {ρ → 1, σx → 1, σϵ → 1, σn → 2, μx → 5};
```

Dealer market: competitive solution with price conjecture (Glosten (1989))

```
R_Rule = R[q_] :→ P[q] q;
P_Conjecture = P[q_] :→ k0 + k1 q;
R[q] /. R_Rule /. P_Conjecture
```

$$q\,(k_0+q\,k_1)$$

```
MarginalCondition = P[q] == ℰPoint[m] /. m -> ∂q (R[q] /. R_Rule /. P_Conjecture)
```

$$P[q]=\frac{\sigma_x^2\left(k_0+2\,q\,k_1+q\,\rho\,\sigma_\epsilon^2\right)+\sigma_\epsilon^2\left(k_0+2\,q\,k_1+\rho^2\,\mu_x\,\sigma_n^2\,\sigma_\epsilon^2\right)}{\sigma_x^2+\sigma_\epsilon^2+\rho^2\,\sigma_n^2\,\sigma_\epsilon^4}$$

```
r = Reduce[
  ∀q (MarginalCondition /. P_Conjecture) && ρ ≠ 0 && σ_ϵ ≠ 0 && σ_x ≠ 0 && σ_n ≠ 0, {k0, k1}]
```

$k_0 == \mu_x \;\&\&\; -\sigma_x^2 - \sigma_\epsilon^2 + \rho^2\,\sigma_n^2\,\sigma_\epsilon^4 \neq 0 \;\&\&$

$k_1 == \dfrac{\rho\,\sigma_x^2\,\sigma_\epsilon^2}{-\sigma_x^2 - \sigma_\epsilon^2 + \rho^2\,\sigma_n^2\,\sigma_\epsilon^4} \;\&\&\; \rho\,\sigma_n\,\sigma_x^3\,\sigma_\epsilon + \rho\,\sigma_n\,\sigma_x\,\sigma_\epsilon^3 + \rho^3\,\sigma_n^3\,\sigma_x\,\sigma_\epsilon^5 \neq 0$

```
k_Rules = ToRules[r[[1]] && r[[3]]]
```

$\left\{k_0 \to \mu_x,\; k_1 \to \dfrac{\rho\,\sigma_x^2\,\sigma_\epsilon^2}{-\sigma_x^2 - \sigma_\epsilon^2 + \rho^2\,\sigma_n^2\,\sigma_\epsilon^4}\right\}$

Sign the derivatives.

```
TableForm[Simplify[{#, D[k1 /. k_Rules, #] > 0} & /@ {σ_x, σ_ϵ, σ_n, ρ},
   {σ_x > 0, σ_n > 0, σ_ϵ > 0, ρ > 0, -σ_x^2 - σ_ϵ^2 + ρ^2 σ_n^2 σ_ϵ^4 > 0}],
  TableHeadings → {{}, {Parameter, "∂_Parameter k1>0?"}}]
```

Parameter	$\partial_{\text{Parameter}} k_1$>0?
σ_x	True
σ_ϵ	False
σ_n	False
ρ	False

```
P_Comp Dealer Rule = P_Conjecture /. k_Rules
```

$P[q\_] :\to \mu_x + \dfrac{\left(\rho\,\sigma_x^2\,\sigma_\epsilon^2\right) q}{-\sigma_x^2 - \sigma_\epsilon^2 + \rho^2\,\sigma_n^2\,\sigma_\epsilon^4}$

Verify that the solution for k_1 agrees with Glosten solution:

```
ρ π_n π_ϵ / (N (2 α - 1)) /. N → ρ^2 π_x + π_n π_ϵ (π_ϵ + π_x) /. α → ρ^2 π_x / (ρ^2 π_x + π_n π_ϵ (π_ϵ + π_x)) /. π_x_ :→ 1/σ_x^2 //
  Simplify
```

$-\dfrac{\rho\,\sigma_x^2\,\sigma_\epsilon^2}{\sigma_x^2 + \sigma_\epsilon^2 - \rho^2\,\sigma_n^2\,\sigma_\epsilon^4}$

```
P[q] /. P_Comp Dealer Rule /. CommonParams
```

$5 + \dfrac{q}{2}$

```
$TextStyle = {FontFamily → "Times", FontSize → 12, LineSpacing → {1, -1}};
```

```
Linestyle = Table[Thickness[.003 x], {x, 1, 3}];
```

```
Plot[Evaluate[P[q] /. P_{Comp Dealer Rule} /. CommonParams], {q, 0, 8},
 AxesOrigin → {0, 5}, PlotStyle → Linestyle, PlotRange → {{0, 9}, {4.9, 7.5}},
 Frame → True, FrameTicks → {Range[0, 10, 1], Range[5, 8, 1], None, None},
 FrameLabel → {"Quantity purchased", "Price and E[x]", None, None},
 BaseStyle → {FontFamily → "Times"}]
```



Limit order market (Glosten (1994))

Glosten Normalizations

```
ρ GetVariance[MVNConditional[d3, x, s]] == 1 // Simplify
```

$$\frac{\rho\,\sigma_x^2\,\sigma_\epsilon^2}{\sigma_x^2+\sigma_\epsilon^2} == 1$$

Glosten suggests: $\rho\,\sigma_{X|S}^2 = 1$; $\sigma_W^2 = \alpha < 1$; $\mathrm{Var}[\mu_{X|S}] = 1 - \alpha$. Check:

$$\text{GlostenNormalizations} = \left\{\rho\ \text{GetVariance}[\text{MVNConditional}[d3, x, s]] == 1,\ \frac{\sigma_x^4}{\sigma_x^2+\sigma_\epsilon^2} == 1-\alpha,\ \sigma_n^2 == \alpha,\ \mu_x == 0\right\}\ //\ \text{Simplify}$$

$$\left\{\frac{\rho\,\sigma_x^2\,\sigma_\epsilon^2}{\sigma_x^2+\sigma_\epsilon^2} == 1,\ \alpha+\frac{\sigma_x^4}{\sigma_x^2+\sigma_\epsilon^2} == 1,\ \alpha == \sigma_n^2,\ \mu_x == 0\right\}$$

```
Eliminate[GlostenNormalizations, α]
```

$$\mu_x == 0 \; \&\& \; \sigma_n^2 == 1 - \frac{\sigma_x^4}{\sigma_x^2 + \sigma_\epsilon^2} \; \&\& \; \rho\, \sigma_x^2\, \sigma_\epsilon^2 == \sigma_x^2 + \sigma_\epsilon^2 \; \&\& \; \rho\, \sigma_x^2 \left(-1 + \frac{\sigma_x^2}{\sigma_x^2 + \sigma_\epsilon^2}\right) == -1$$

solve for some numerical values:

```
Solve[GlostenNormalizations /. σx → 1 /. σϵ → 1.2] // N // TableForm
```

$\rho \to 1.69444 \quad \mu_x \to 0. \quad \alpha \to 0.590164 \quad \sigma_n \to -0.768221$
$\rho \to 1.69444 \quad \mu_x \to 0. \quad \alpha \to 0.590164 \quad \sigma_n \to 0.768221$

Verify that $\omega \equiv \mu_{x\,|\,s} - \rho\sigma^2\, W \Rightarrow \mathrm{Var}[\omega] = 1$

```
Solve[Append[GlostenNormalizations,
  GetVariance[MakeLinearForm[d3, {ω}]] == σω^2], σω]
```

$\{\{\sigma_\omega \to -1\}, \{\sigma_\omega \to 1\}\}$

Verify that with `GlostenNormalizations`, $\mathcal{E}C \equiv E[X \mid \omega = w] = (1 - \alpha)\, w$:

```
Solve[Append[GlostenNormalizations,
  GetMean[MVNConditional[d3, x, ω]] == ℰC], ℰC, σn]
```

$\{\{\mathcal{E}C \to \omega - \alpha\, \omega\}\}$

□ *General case (not imposing Glosten normalizations)*

```
d3
```

$$\begin{pmatrix} x \\ s \\ n \\ \omega \end{pmatrix} \sim \mathcal{N}\left(\begin{pmatrix} \mu_x \\ \mu_x \\ 0 \\ \mu_x \end{pmatrix}, \begin{pmatrix} \sigma_x^2 & \sigma_x^2 & 0 & \frac{\sigma_x^4}{\sigma_x^2+\sigma_\epsilon^2} \\ \sigma_x^2 & \sigma_x^2 + \sigma_\epsilon^2 & 0 & \sigma_x^2 \\ 0 & 0 & \sigma_n^2 & -\frac{\rho\,\sigma_n^2\,\sigma_x^2\,\sigma_\epsilon^2}{\sigma_x^2+\sigma_\epsilon^2} \\ \frac{\sigma_x^4}{\sigma_x^2+\sigma_\epsilon^2} & \sigma_x^2 & -\frac{\rho\,\sigma_n^2\,\sigma_x^2\,\sigma_\epsilon^2}{\sigma_x^2+\sigma_\epsilon^2} & \frac{\sigma_x^4\left(\sigma_x^2+\sigma_\epsilon^2+\rho^2\,\sigma_n^2\,\sigma_\epsilon^4\right)}{\left(\sigma_x^2+\sigma_\epsilon^2\right)^2} \end{pmatrix}\right)$$

```
d4 = MVNConditional[d3, x, s]
```

$$x \sim \mathcal{N}\left(\mu_x + \frac{(s - \mu_x)\,\sigma_x^2}{\sigma_x^2 + \sigma_\epsilon^2}, \; \sigma_x^2 - \frac{\sigma_x^4}{\sigma_x^2 + \sigma_\epsilon^2}\right)$$

```
vv = GetVariance[d4]
```

$$\sigma_x^2 - \frac{\sigma_x^4}{\sigma_x^2 + \sigma_\epsilon^2}$$

```
V = GetMeanTrunc[d3, x, ω, m + q ρ vv] // FullSimplify
```

$$\mu_x - \frac{\sigma_x^4\, \phi\left[\frac{m-\mu_x+\frac{q\rho\sigma_x^2\sigma_\epsilon^2}{\sigma_x^2+\sigma_\epsilon^2}}{\sqrt{\frac{\sigma_x^4\left(\sigma_x^2+\sigma_\epsilon^2+\rho^2\sigma_n^2\sigma_\epsilon^4\right)}{\left(\sigma_x^2+\sigma_\epsilon^2\right)^2}}}\right]}{\left(\sigma_x^2+\sigma_\epsilon^2\right)\sqrt{\frac{\sigma_x^4\left(\sigma_x^2+\sigma_\epsilon^2+\rho^2\sigma_n^2\sigma_\epsilon^4\right)}{\left(\sigma_x^2+\sigma_\epsilon^2\right)^2}}\left(-1+\Phi\left[\frac{m-\mu_x+\frac{q\rho\sigma_x^2\sigma_\epsilon^2}{\sigma_x^2+\sigma_\epsilon^2}}{\sqrt{\frac{\sigma_x^4\left(\sigma_x^2+\sigma_\epsilon^2+\rho^2\sigma_n^2\sigma_\epsilon^4\right)}{\left(\sigma_x^2+\sigma_\epsilon^2\right)^2}}}\right]\right)}$$

□ *Check: Compare above with Glosten (1994) top of p. 1141.*

Verify that under the `GlostenNormalizations`, the argument of ϕ reduces to $m + q$

```
ϕArg = First[Cases[V, ϕ[a_] → a, ∞]]
```

$$\frac{m-\mu_x+\frac{q\rho\sigma_x^2\sigma_\epsilon^2}{\sigma_x^2+\sigma_\epsilon^2}}{\sqrt{\frac{\sigma_x^4\left(\sigma_x^2+\sigma_\epsilon^2+\rho^2\sigma_n^2\sigma_\epsilon^4\right)}{\left(\sigma_x^2+\sigma_\epsilon^2\right)^2}}}$$

```
Simplify[Solve[Append[GlostenNormalizations, k == ϕArg], k, σn], {m > 0, q > 0}]
```

$\{\{k \to -m - q\}, \{k \to m + q\}\}$

Verify that under the `GlostenNormalizations`, the coefficient of $\phi / (1 - \Phi)$ reduces to $1 - \alpha$

```
factor = V /. ϕ[_] → 1 /. Φ[_] → 0
```

$$\mu_x + \frac{\sigma_x^4}{\left(\sigma_x^2+\sigma_\epsilon^2\right)\sqrt{\frac{\sigma_x^4\left(\sigma_x^2+\sigma_\epsilon^2+\rho^2\sigma_n^2\sigma_\epsilon^4\right)}{\left(\sigma_x^2+\sigma_\epsilon^2\right)^2}}}$$

```
Simplify[Solve[Append[GlostenNormalizations, k == factor], k, σn], {0 < α < 1}]
```

$\{\{k \to -1 + \alpha\}, \{k \to 1 - \alpha\}\}$

□ *Numerical examples with common parameters*

Does competitive solution exist?

```
0 < ρ^2 σn^2 σϵ^2 - (σx^2 + σϵ^2) /. CommonParams
```

```
True
```

In the discriminating limit order book, the price is equal to the upper-tail expectation:

```
equu = p == (V /. CommonParams /. m → p /. Normal_Rules)
```

$$p == 5 - \frac{e^{-\frac{1}{3}\left(-5+p+\frac{q}{2}\right)^2}}{2\sqrt{3\pi}\left(-1+\frac{1}{2}\left(1+\text{Erf}\left[\frac{-5+p+\frac{q}{2}}{\sqrt{3}}\right]\right)\right)}$$

For example, at $q = 8$, the price is:

```
FindRoot[equ /. q → 8, {p, 1}]
```

$\{p \to 7.11436\}$

By solving this for a number of q and interpolating, we may approximate the supply curve.

```
P_Discrim = Interpolation[
   Table[{q, p /. FindRoot[equ, {p, 1}, AccuracyGoal → 6]}, {q, 0, 8, .05}]];
```

Plot the supply curve along with $E\left[X \mid \omega = m + q\,\rho\,\sigma^2_{X|S}\right]$, i.e., when the incoming quantity just executes the order.

```
Table[V /. m → P_Discrim[q] /. CommonParams /. Normal_Rules, {q, 1, 6}]
```

{5.58587, 5.77346, 5.97698, 6.192, 6.41522, 6.64434}

```
Table[P_Discrim[q] /. CommonParams, {q, 1, 6}]
```

{5.58587, 5.77346, 5.97698, 6.192, 6.41522, 6.64434}

```
Plot[Evaluate[{P_Discrim[q], ℰPoint[m] /. m → P_Discrim[q]} /. CommonParams],
 {q, 0, 8}, AxesOrigin → {0, 5}, PlotRange → {{0, 9}, {4.9, 7.5}}, Frame → True,
 FrameLabel → {"Quantity purchased, q ", "Price and E[X]", None, None},
 FrameTicks → {Range[0, 10, 1], Range[5, 8, 1], None, None},
 Background → GrayLevel[1], BaseStyle → {FontFamily → "Times"}]
```